// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```solidity
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```solidity
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it. We recommend using {AccessControlDefaultAdminRules}
 * to enforce additional security measures for this role.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(account),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleGranted} event.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleRevoked} event.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     *
     * May emit a {RoleRevoked} event.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * May emit a {RoleGranted} event.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleGranted} event.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleRevoked} event.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/interfaces/IERC2981.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (interfaces/IERC2981.sol)

pragma solidity ^0.8.0;

import "../utils/introspection/IERC165.sol";

/**
 * @dev Interface for the NFT Royalty Standard.
 *
 * A standardized way to retrieve royalty payment information for non-fungible tokens (NFTs) to enable universal
 * support for royalty payments across all NFT marketplaces and ecosystem participants.
 *
 * _Available since v4.5._
 */
interface IERC2981 is IERC165 {
    /**
     * @dev Returns how much royalty is owed and to whom, based on a sale price that may be denominated in any unit of
     * exchange. The royalty amount is denominated and should be paid in that same unit of exchange.
     */
    function royaltyInfo(
        uint256 tokenId,
        uint256 salePrice
    ) external view returns (address receiver, uint256 royaltyAmount);
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/extensions/IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    /**
     * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    /**
     * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
     * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
     */
    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    /**
     * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 oldAllowance = token.allowance(address(this), spender);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance + value));
    }

    /**
     * @dev Decrease the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance - value));
        }
    }

    /**
     * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful. Compatible with tokens that require the approval to be set to
     * 0 before setting it to a non-zero value.
     */
    function forceApprove(IERC20 token, address spender, uint256 value) internal {
        bytes memory approvalCall = abi.encodeWithSelector(token.approve.selector, spender, value);

        if (!_callOptionalReturnBool(token, approvalCall)) {
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, 0));
            _callOptionalReturn(token, approvalCall);
        }
    }

    /**
     * @dev Use a ERC-2612 signature to set the `owner` approval toward `spender` on `token`.
     * Revert on invalid signature.
     */
    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        require(returndata.length == 0 || abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     *
     * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
     */
    function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
        // and not revert is the subcall reverts.

        (bool success, bytes memory returndata) = address(token).call(data);
        return
            success && (returndata.length == 0 || abi.decode(returndata, (bool))) && Address.isContract(address(token));
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _ownerOf(tokenId);
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner or approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(address from, address to, uint256 tokenId) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes memory data) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(address from, address to, uint256 tokenId, bytes memory data) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns the owner of the `tokenId`. Does NOT revert if token doesn't exist
     */
    function _ownerOf(uint256 tokenId) internal view virtual returns (address) {
        return _owners[tokenId];
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _ownerOf(tokenId) != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(address to, uint256 tokenId, bytes memory data) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId, 1);

        // Check that tokenId was not minted by `_beforeTokenTransfer` hook
        require(!_exists(tokenId), "ERC721: token already minted");

        unchecked {
            // Will not overflow unless all 2**256 token ids are minted to the same owner.
            // Given that tokens are minted one by one, it is impossible in practice that
            // this ever happens. Might change if we allow batch minting.
            // The ERC fails to describe this case.
            _balances[to] += 1;
        }

        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId, 1);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     * This is an internal function that does not check if the sender is authorized to operate on the token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId, 1);

        // Update ownership in case tokenId was transferred by `_beforeTokenTransfer` hook
        owner = ERC721.ownerOf(tokenId);

        // Clear approvals
        delete _tokenApprovals[tokenId];

        unchecked {
            // Cannot overflow, as that would require more tokens to be burned/transferred
            // out than the owner initially received through minting and transferring in.
            _balances[owner] -= 1;
        }
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId, 1);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(address from, address to, uint256 tokenId) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId, 1);

        // Check that tokenId was not transferred by `_beforeTokenTransfer` hook
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");

        // Clear approvals from the previous owner
        delete _tokenApprovals[tokenId];

        unchecked {
            // `_balances[from]` cannot overflow for the same reason as described in `_burn`:
            // `from`'s balance is the number of token held, which is at least one before the current
            // transfer.
            // `_balances[to]` could overflow in the conditions described in `_mint`. That would require
            // all 2**256 token ids to be minted, which in practice is impossible.
            _balances[from] -= 1;
            _balances[to] += 1;
        }
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId, 1);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(address owner, address operator, bool approved) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens will be transferred to `to`.
     * - When `from` is zero, the tokens will be minted for `to`.
     * - When `to` is zero, ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 firstTokenId, uint256 batchSize) internal virtual {}

    /**
     * @dev Hook that is called after any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens were transferred to `to`.
     * - When `from` is zero, the tokens were minted for `to`.
     * - When `to` is zero, ``from``'s tokens were burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 firstTokenId, uint256 batchSize) internal virtual {}

    /**
     * @dev Unsafe write access to the balances, used by extensions that "mint" tokens using an {ownerOf} override.
     *
     * WARNING: Anyone calling this MUST ensure that the balances remain consistent with the ownership. The invariant
     * being that for any address `a` the value returned by `balanceOf(a)` must be equal to the number of tokens such
     * that `ownerOf(tokenId)` is `a`.
     */
    // solhint-disable-next-line func-name-mixedcase
    function __unsafe_increaseBalance(address account, uint256 amount) internal {
        _balances[account] += amount;
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721Royalty.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/extensions/ERC721Royalty.sol)

pragma solidity ^0.8.0;

import "../ERC721.sol";
import "../../common/ERC2981.sol";
import "../../../utils/introspection/ERC165.sol";

/**
 * @dev Extension of ERC721 with the ERC2981 NFT Royalty Standard, a standardized way to retrieve royalty payment
 * information.
 *
 * Royalty information can be specified globally for all token ids via {ERC2981-_setDefaultRoyalty}, and/or individually for
 * specific token ids via {ERC2981-_setTokenRoyalty}. The latter takes precedence over the first.
 *
 * IMPORTANT: ERC-2981 only specifies a way to signal royalty information and does not enforce its payment. See
 * https://eips.ethereum.org/EIPS/eip-2981#optional-royalty-payments[Rationale] in the EIP. Marketplaces are expected to
 * voluntarily pay royalties together with sales, but note that this standard is not yet widely supported.
 *
 * _Available since v4.5._
 */
abstract contract ERC721Royalty is ERC2981, ERC721 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC721, ERC2981) returns (bool) {
        return super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {ERC721-_burn}. This override additionally clears the royalty information for the token.
     */
    function _burn(uint256 tokenId) internal virtual override {
        super._burn(tokenId);
        _resetTokenRoyalty(tokenId);
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/common/ERC2981.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/common/ERC2981.sol)

pragma solidity ^0.8.0;

import "../../interfaces/IERC2981.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of the NFT Royalty Standard, a standardized way to retrieve royalty payment information.
 *
 * Royalty information can be specified globally for all token ids via {_setDefaultRoyalty}, and/or individually for
 * specific token ids via {_setTokenRoyalty}. The latter takes precedence over the first.
 *
 * Royalty is specified as a fraction of sale price. {_feeDenominator} is overridable but defaults to 10000, meaning the
 * fee is specified in basis points by default.
 *
 * IMPORTANT: ERC-2981 only specifies a way to signal royalty information and does not enforce its payment. See
 * https://eips.ethereum.org/EIPS/eip-2981#optional-royalty-payments[Rationale] in the EIP. Marketplaces are expected to
 * voluntarily pay royalties together with sales, but note that this standard is not yet widely supported.
 *
 * _Available since v4.5._
 */
abstract contract ERC2981 is IERC2981, ERC165 {
    struct RoyaltyInfo {
        address receiver;
        uint96 royaltyFraction;
    }

    RoyaltyInfo private _defaultRoyaltyInfo;
    mapping(uint256 => RoyaltyInfo) private _tokenRoyaltyInfo;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC165) returns (bool) {
        return interfaceId == type(IERC2981).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @inheritdoc IERC2981
     */
    function royaltyInfo(uint256 tokenId, uint256 salePrice) public view virtual override returns (address, uint256) {
        RoyaltyInfo memory royalty = _tokenRoyaltyInfo[tokenId];

        if (royalty.receiver == address(0)) {
            royalty = _defaultRoyaltyInfo;
        }

        uint256 royaltyAmount = (salePrice * royalty.royaltyFraction) / _feeDenominator();

        return (royalty.receiver, royaltyAmount);
    }

    /**
     * @dev The denominator with which to interpret the fee set in {_setTokenRoyalty} and {_setDefaultRoyalty} as a
     * fraction of the sale price. Defaults to 10000 so fees are expressed in basis points, but may be customized by an
     * override.
     */
    function _feeDenominator() internal pure virtual returns (uint96) {
        return 10000;
    }

    /**
     * @dev Sets the royalty information that all ids in this contract will default to.
     *
     * Requirements:
     *
     * - `receiver` cannot be the zero address.
     * - `feeNumerator` cannot be greater than the fee denominator.
     */
    function _setDefaultRoyalty(address receiver, uint96 feeNumerator) internal virtual {
        require(feeNumerator <= _feeDenominator(), "ERC2981: royalty fee will exceed salePrice");
        require(receiver != address(0), "ERC2981: invalid receiver");

        _defaultRoyaltyInfo = RoyaltyInfo(receiver, feeNumerator);
    }

    /**
     * @dev Removes default royalty information.
     */
    function _deleteDefaultRoyalty() internal virtual {
        delete _defaultRoyaltyInfo;
    }

    /**
     * @dev Sets the royalty information for a specific token id, overriding the global default.
     *
     * Requirements:
     *
     * - `receiver` cannot be the zero address.
     * - `feeNumerator` cannot be greater than the fee denominator.
     */
    function _setTokenRoyalty(uint256 tokenId, address receiver, uint96 feeNumerator) internal virtual {
        require(feeNumerator <= _feeDenominator(), "ERC2981: royalty fee will exceed salePrice");
        require(receiver != address(0), "ERC2981: Invalid parameters");

        _tokenRoyaltyInfo[tokenId] = RoyaltyInfo(receiver, feeNumerator);
    }

    /**
     * @dev Resets royalty information for the token id back to the global default.
     */
    function _resetTokenRoyalty(uint256 tokenId) internal virtual {
        delete _tokenRoyaltyInfo[tokenId];
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Counters.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Counters.sol)

pragma solidity ^0.8.0;

/**
 * @title Counters
 * @author Matt Condon (@shrugs)
 * @dev Provides counters that can only be incremented, decremented or reset. This can be used e.g. to track the number
 * of elements in a mapping, issuing ERC721 ids, or counting request ids.
 *
 * Include with `using Counters for Counters.Counter;`
 */
library Counters {
    struct Counter {
        // This variable should never be directly accessed by users of the library: interactions must be restricted to
        // the library's function. As of Solidity v0.5.2, this cannot be enforced, though there is a proposal to add
        // this feature: see https://github.com/ethereum/solidity/issues/4637
        uint256 _value; // default: 0
    }

    function current(Counter storage counter) internal view returns (uint256) {
        return counter._value;
    }

    function increment(Counter storage counter) internal {
        unchecked {
            counter._value += 1;
        }
    }

    function decrement(Counter storage counter) internal {
        uint256 value = counter._value;
        require(value > 0, "Counter: decrement overflow");
        unchecked {
            counter._value = value - 1;
        }
    }

    function reset(Counter storage counter) internal {
        counter._value = 0;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";
import "./math/SignedMath.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `int256` to its ASCII `string` decimal representation.
     */
    function toString(int256 value) internal pure returns (string memory) {
        return string(abi.encodePacked(value < 0 ? "-" : "", toString(SignedMath.abs(value))));
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }

    /**
     * @dev Returns true if the two strings are equal.
     */
    function equal(string memory a, string memory b) internal pure returns (bool) {
        return keccak256(bytes(a)) == keccak256(bytes(b));
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                // The surrounding unchecked block does not change this fact.
                // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1, "Math: mulDiv overflow");

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10 ** 64) {
                value /= 10 ** 64;
                result += 64;
            }
            if (value >= 10 ** 32) {
                value /= 10 ** 32;
                result += 32;
            }
            if (value >= 10 ** 16) {
                value /= 10 ** 16;
                result += 16;
            }
            if (value >= 10 ** 8) {
                value /= 10 ** 8;
                result += 8;
            }
            if (value >= 10 ** 4) {
                value /= 10 ** 4;
                result += 4;
            }
            if (value >= 10 ** 2) {
                value /= 10 ** 2;
                result += 2;
            }
            if (value >= 10 ** 1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/math/SafeCast.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/SafeCast.sol)
// This file was procedurally generated from scripts/generate/templates/SafeCast.js.

pragma solidity ^0.8.0;

/**
 * @dev Wrappers over Solidity's uintXX/intXX casting operators with added overflow
 * checks.
 *
 * Downcasting from uint256/int256 in Solidity does not revert on overflow. This can
 * easily result in undesired exploitation or bugs, since developers usually
 * assume that overflows raise errors. `SafeCast` restores this intuition by
 * reverting the transaction when such an operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 *
 * Can be combined with {SafeMath} and {SignedSafeMath} to extend it to smaller types, by performing
 * all math on `uint256` and `int256` and then downcasting.
 */
library SafeCast {
    /**
     * @dev Returns the downcasted uint248 from uint256, reverting on
     * overflow (when the input is greater than largest uint248).
     *
     * Counterpart to Solidity's `uint248` operator.
     *
     * Requirements:
     *
     * - input must fit into 248 bits
     *
     * _Available since v4.7._
     */
    function toUint248(uint256 value) internal pure returns (uint248) {
        require(value <= type(uint248).max, "SafeCast: value doesn't fit in 248 bits");
        return uint248(value);
    }

    /**
     * @dev Returns the downcasted uint240 from uint256, reverting on
     * overflow (when the input is greater than largest uint240).
     *
     * Counterpart to Solidity's `uint240` operator.
     *
     * Requirements:
     *
     * - input must fit into 240 bits
     *
     * _Available since v4.7._
     */
    function toUint240(uint256 value) internal pure returns (uint240) {
        require(value <= type(uint240).max, "SafeCast: value doesn't fit in 240 bits");
        return uint240(value);
    }

    /**
     * @dev Returns the downcasted uint232 from uint256, reverting on
     * overflow (when the input is greater than largest uint232).
     *
     * Counterpart to Solidity's `uint232` operator.
     *
     * Requirements:
     *
     * - input must fit into 232 bits
     *
     * _Available since v4.7._
     */
    function toUint232(uint256 value) internal pure returns (uint232) {
        require(value <= type(uint232).max, "SafeCast: value doesn't fit in 232 bits");
        return uint232(value);
    }

    /**
     * @dev Returns the downcasted uint224 from uint256, reverting on
     * overflow (when the input is greater than largest uint224).
     *
     * Counterpart to Solidity's `uint224` operator.
     *
     * Requirements:
     *
     * - input must fit into 224 bits
     *
     * _Available since v4.2._
     */
    function toUint224(uint256 value) internal pure returns (uint224) {
        require(value <= type(uint224).max, "SafeCast: value doesn't fit in 224 bits");
        return uint224(value);
    }

    /**
     * @dev Returns the downcasted uint216 from uint256, reverting on
     * overflow (when the input is greater than largest uint216).
     *
     * Counterpart to Solidity's `uint216` operator.
     *
     * Requirements:
     *
     * - input must fit into 216 bits
     *
     * _Available since v4.7._
     */
    function toUint216(uint256 value) internal pure returns (uint216) {
        require(value <= type(uint216).max, "SafeCast: value doesn't fit in 216 bits");
        return uint216(value);
    }

    /**
     * @dev Returns the downcasted uint208 from uint256, reverting on
     * overflow (when the input is greater than largest uint208).
     *
     * Counterpart to Solidity's `uint208` operator.
     *
     * Requirements:
     *
     * - input must fit into 208 bits
     *
     * _Available since v4.7._
     */
    function toUint208(uint256 value) internal pure returns (uint208) {
        require(value <= type(uint208).max, "SafeCast: value doesn't fit in 208 bits");
        return uint208(value);
    }

    /**
     * @dev Returns the downcasted uint200 from uint256, reverting on
     * overflow (when the input is greater than largest uint200).
     *
     * Counterpart to Solidity's `uint200` operator.
     *
     * Requirements:
     *
     * - input must fit into 200 bits
     *
     * _Available since v4.7._
     */
    function toUint200(uint256 value) internal pure returns (uint200) {
        require(value <= type(uint200).max, "SafeCast: value doesn't fit in 200 bits");
        return uint200(value);
    }

    /**
     * @dev Returns the downcasted uint192 from uint256, reverting on
     * overflow (when the input is greater than largest uint192).
     *
     * Counterpart to Solidity's `uint192` operator.
     *
     * Requirements:
     *
     * - input must fit into 192 bits
     *
     * _Available since v4.7._
     */
    function toUint192(uint256 value) internal pure returns (uint192) {
        require(value <= type(uint192).max, "SafeCast: value doesn't fit in 192 bits");
        return uint192(value);
    }

    /**
     * @dev Returns the downcasted uint184 from uint256, reverting on
     * overflow (when the input is greater than largest uint184).
     *
     * Counterpart to Solidity's `uint184` operator.
     *
     * Requirements:
     *
     * - input must fit into 184 bits
     *
     * _Available since v4.7._
     */
    function toUint184(uint256 value) internal pure returns (uint184) {
        require(value <= type(uint184).max, "SafeCast: value doesn't fit in 184 bits");
        return uint184(value);
    }

    /**
     * @dev Returns the downcasted uint176 from uint256, reverting on
     * overflow (when the input is greater than largest uint176).
     *
     * Counterpart to Solidity's `uint176` operator.
     *
     * Requirements:
     *
     * - input must fit into 176 bits
     *
     * _Available since v4.7._
     */
    function toUint176(uint256 value) internal pure returns (uint176) {
        require(value <= type(uint176).max, "SafeCast: value doesn't fit in 176 bits");
        return uint176(value);
    }

    /**
     * @dev Returns the downcasted uint168 from uint256, reverting on
     * overflow (when the input is greater than largest uint168).
     *
     * Counterpart to Solidity's `uint168` operator.
     *
     * Requirements:
     *
     * - input must fit into 168 bits
     *
     * _Available since v4.7._
     */
    function toUint168(uint256 value) internal pure returns (uint168) {
        require(value <= type(uint168).max, "SafeCast: value doesn't fit in 168 bits");
        return uint168(value);
    }

    /**
     * @dev Returns the downcasted uint160 from uint256, reverting on
     * overflow (when the input is greater than largest uint160).
     *
     * Counterpart to Solidity's `uint160` operator.
     *
     * Requirements:
     *
     * - input must fit into 160 bits
     *
     * _Available since v4.7._
     */
    function toUint160(uint256 value) internal pure returns (uint160) {
        require(value <= type(uint160).max, "SafeCast: value doesn't fit in 160 bits");
        return uint160(value);
    }

    /**
     * @dev Returns the downcasted uint152 from uint256, reverting on
     * overflow (when the input is greater than largest uint152).
     *
     * Counterpart to Solidity's `uint152` operator.
     *
     * Requirements:
     *
     * - input must fit into 152 bits
     *
     * _Available since v4.7._
     */
    function toUint152(uint256 value) internal pure returns (uint152) {
        require(value <= type(uint152).max, "SafeCast: value doesn't fit in 152 bits");
        return uint152(value);
    }

    /**
     * @dev Returns the downcasted uint144 from uint256, reverting on
     * overflow (when the input is greater than largest uint144).
     *
     * Counterpart to Solidity's `uint144` operator.
     *
     * Requirements:
     *
     * - input must fit into 144 bits
     *
     * _Available since v4.7._
     */
    function toUint144(uint256 value) internal pure returns (uint144) {
        require(value <= type(uint144).max, "SafeCast: value doesn't fit in 144 bits");
        return uint144(value);
    }

    /**
     * @dev Returns the downcasted uint136 from uint256, reverting on
     * overflow (when the input is greater than largest uint136).
     *
     * Counterpart to Solidity's `uint136` operator.
     *
     * Requirements:
     *
     * - input must fit into 136 bits
     *
     * _Available since v4.7._
     */
    function toUint136(uint256 value) internal pure returns (uint136) {
        require(value <= type(uint136).max, "SafeCast: value doesn't fit in 136 bits");
        return uint136(value);
    }

    /**
     * @dev Returns the downcasted uint128 from uint256, reverting on
     * overflow (when the input is greater than largest uint128).
     *
     * Counterpart to Solidity's `uint128` operator.
     *
     * Requirements:
     *
     * - input must fit into 128 bits
     *
     * _Available since v2.5._
     */
    function toUint128(uint256 value) internal pure returns (uint128) {
        require(value <= type(uint128).max, "SafeCast: value doesn't fit in 128 bits");
        return uint128(value);
    }

    /**
     * @dev Returns the downcasted uint120 from uint256, reverting on
     * overflow (when the input is greater than largest uint120).
     *
     * Counterpart to Solidity's `uint120` operator.
     *
     * Requirements:
     *
     * - input must fit into 120 bits
     *
     * _Available since v4.7._
     */
    function toUint120(uint256 value) internal pure returns (uint120) {
        require(value <= type(uint120).max, "SafeCast: value doesn't fit in 120 bits");
        return uint120(value);
    }

    /**
     * @dev Returns the downcasted uint112 from uint256, reverting on
     * overflow (when the input is greater than largest uint112).
     *
     * Counterpart to Solidity's `uint112` operator.
     *
     * Requirements:
     *
     * - input must fit into 112 bits
     *
     * _Available since v4.7._
     */
    function toUint112(uint256 value) internal pure returns (uint112) {
        require(value <= type(uint112).max, "SafeCast: value doesn't fit in 112 bits");
        return uint112(value);
    }

    /**
     * @dev Returns the downcasted uint104 from uint256, reverting on
     * overflow (when the input is greater than largest uint104).
     *
     * Counterpart to Solidity's `uint104` operator.
     *
     * Requirements:
     *
     * - input must fit into 104 bits
     *
     * _Available since v4.7._
     */
    function toUint104(uint256 value) internal pure returns (uint104) {
        require(value <= type(uint104).max, "SafeCast: value doesn't fit in 104 bits");
        return uint104(value);
    }

    /**
     * @dev Returns the downcasted uint96 from uint256, reverting on
     * overflow (when the input is greater than largest uint96).
     *
     * Counterpart to Solidity's `uint96` operator.
     *
     * Requirements:
     *
     * - input must fit into 96 bits
     *
     * _Available since v4.2._
     */
    function toUint96(uint256 value) internal pure returns (uint96) {
        require(value <= type(uint96).max, "SafeCast: value doesn't fit in 96 bits");
        return uint96(value);
    }

    /**
     * @dev Returns the downcasted uint88 from uint256, reverting on
     * overflow (when the input is greater than largest uint88).
     *
     * Counterpart to Solidity's `uint88` operator.
     *
     * Requirements:
     *
     * - input must fit into 88 bits
     *
     * _Available since v4.7._
     */
    function toUint88(uint256 value) internal pure returns (uint88) {
        require(value <= type(uint88).max, "SafeCast: value doesn't fit in 88 bits");
        return uint88(value);
    }

    /**
     * @dev Returns the downcasted uint80 from uint256, reverting on
     * overflow (when the input is greater than largest uint80).
     *
     * Counterpart to Solidity's `uint80` operator.
     *
     * Requirements:
     *
     * - input must fit into 80 bits
     *
     * _Available since v4.7._
     */
    function toUint80(uint256 value) internal pure returns (uint80) {
        require(value <= type(uint80).max, "SafeCast: value doesn't fit in 80 bits");
        return uint80(value);
    }

    /**
     * @dev Returns the downcasted uint72 from uint256, reverting on
     * overflow (when the input is greater than largest uint72).
     *
     * Counterpart to Solidity's `uint72` operator.
     *
     * Requirements:
     *
     * - input must fit into 72 bits
     *
     * _Available since v4.7._
     */
    function toUint72(uint256 value) internal pure returns (uint72) {
        require(value <= type(uint72).max, "SafeCast: value doesn't fit in 72 bits");
        return uint72(value);
    }

    /**
     * @dev Returns the downcasted uint64 from uint256, reverting on
     * overflow (when the input is greater than largest uint64).
     *
     * Counterpart to Solidity's `uint64` operator.
     *
     * Requirements:
     *
     * - input must fit into 64 bits
     *
     * _Available since v2.5._
     */
    function toUint64(uint256 value) internal pure returns (uint64) {
        require(value <= type(uint64).max, "SafeCast: value doesn't fit in 64 bits");
        return uint64(value);
    }

    /**
     * @dev Returns the downcasted uint56 from uint256, reverting on
     * overflow (when the input is greater than largest uint56).
     *
     * Counterpart to Solidity's `uint56` operator.
     *
     * Requirements:
     *
     * - input must fit into 56 bits
     *
     * _Available since v4.7._
     */
    function toUint56(uint256 value) internal pure returns (uint56) {
        require(value <= type(uint56).max, "SafeCast: value doesn't fit in 56 bits");
        return uint56(value);
    }

    /**
     * @dev Returns the downcasted uint48 from uint256, reverting on
     * overflow (when the input is greater than largest uint48).
     *
     * Counterpart to Solidity's `uint48` operator.
     *
     * Requirements:
     *
     * - input must fit into 48 bits
     *
     * _Available since v4.7._
     */
    function toUint48(uint256 value) internal pure returns (uint48) {
        require(value <= type(uint48).max, "SafeCast: value doesn't fit in 48 bits");
        return uint48(value);
    }

    /**
     * @dev Returns the downcasted uint40 from uint256, reverting on
     * overflow (when the input is greater than largest uint40).
     *
     * Counterpart to Solidity's `uint40` operator.
     *
     * Requirements:
     *
     * - input must fit into 40 bits
     *
     * _Available since v4.7._
     */
    function toUint40(uint256 value) internal pure returns (uint40) {
        require(value <= type(uint40).max, "SafeCast: value doesn't fit in 40 bits");
        return uint40(value);
    }

    /**
     * @dev Returns the downcasted uint32 from uint256, reverting on
     * overflow (when the input is greater than largest uint32).
     *
     * Counterpart to Solidity's `uint32` operator.
     *
     * Requirements:
     *
     * - input must fit into 32 bits
     *
     * _Available since v2.5._
     */
    function toUint32(uint256 value) internal pure returns (uint32) {
        require(value <= type(uint32).max, "SafeCast: value doesn't fit in 32 bits");
        return uint32(value);
    }

    /**
     * @dev Returns the downcasted uint24 from uint256, reverting on
     * overflow (when the input is greater than largest uint24).
     *
     * Counterpart to Solidity's `uint24` operator.
     *
     * Requirements:
     *
     * - input must fit into 24 bits
     *
     * _Available since v4.7._
     */
    function toUint24(uint256 value) internal pure returns (uint24) {
        require(value <= type(uint24).max, "SafeCast: value doesn't fit in 24 bits");
        return uint24(value);
    }

    /**
     * @dev Returns the downcasted uint16 from uint256, reverting on
     * overflow (when the input is greater than largest uint16).
     *
     * Counterpart to Solidity's `uint16` operator.
     *
     * Requirements:
     *
     * - input must fit into 16 bits
     *
     * _Available since v2.5._
     */
    function toUint16(uint256 value) internal pure returns (uint16) {
        require(value <= type(uint16).max, "SafeCast: value doesn't fit in 16 bits");
        return uint16(value);
    }

    /**
     * @dev Returns the downcasted uint8 from uint256, reverting on
     * overflow (when the input is greater than largest uint8).
     *
     * Counterpart to Solidity's `uint8` operator.
     *
     * Requirements:
     *
     * - input must fit into 8 bits
     *
     * _Available since v2.5._
     */
    function toUint8(uint256 value) internal pure returns (uint8) {
        require(value <= type(uint8).max, "SafeCast: value doesn't fit in 8 bits");
        return uint8(value);
    }

    /**
     * @dev Converts a signed int256 into an unsigned uint256.
     *
     * Requirements:
     *
     * - input must be greater than or equal to 0.
     *
     * _Available since v3.0._
     */
    function toUint256(int256 value) internal pure returns (uint256) {
        require(value >= 0, "SafeCast: value must be positive");
        return uint256(value);
    }

    /**
     * @dev Returns the downcasted int248 from int256, reverting on
     * overflow (when the input is less than smallest int248 or
     * greater than largest int248).
     *
     * Counterpart to Solidity's `int248` operator.
     *
     * Requirements:
     *
     * - input must fit into 248 bits
     *
     * _Available since v4.7._
     */
    function toInt248(int256 value) internal pure returns (int248 downcasted) {
        downcasted = int248(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 248 bits");
    }

    /**
     * @dev Returns the downcasted int240 from int256, reverting on
     * overflow (when the input is less than smallest int240 or
     * greater than largest int240).
     *
     * Counterpart to Solidity's `int240` operator.
     *
     * Requirements:
     *
     * - input must fit into 240 bits
     *
     * _Available since v4.7._
     */
    function toInt240(int256 value) internal pure returns (int240 downcasted) {
        downcasted = int240(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 240 bits");
    }

    /**
     * @dev Returns the downcasted int232 from int256, reverting on
     * overflow (when the input is less than smallest int232 or
     * greater than largest int232).
     *
     * Counterpart to Solidity's `int232` operator.
     *
     * Requirements:
     *
     * - input must fit into 232 bits
     *
     * _Available since v4.7._
     */
    function toInt232(int256 value) internal pure returns (int232 downcasted) {
        downcasted = int232(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 232 bits");
    }

    /**
     * @dev Returns the downcasted int224 from int256, reverting on
     * overflow (when the input is less than smallest int224 or
     * greater than largest int224).
     *
     * Counterpart to Solidity's `int224` operator.
     *
     * Requirements:
     *
     * - input must fit into 224 bits
     *
     * _Available since v4.7._
     */
    function toInt224(int256 value) internal pure returns (int224 downcasted) {
        downcasted = int224(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 224 bits");
    }

    /**
     * @dev Returns the downcasted int216 from int256, reverting on
     * overflow (when the input is less than smallest int216 or
     * greater than largest int216).
     *
     * Counterpart to Solidity's `int216` operator.
     *
     * Requirements:
     *
     * - input must fit into 216 bits
     *
     * _Available since v4.7._
     */
    function toInt216(int256 value) internal pure returns (int216 downcasted) {
        downcasted = int216(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 216 bits");
    }

    /**
     * @dev Returns the downcasted int208 from int256, reverting on
     * overflow (when the input is less than smallest int208 or
     * greater than largest int208).
     *
     * Counterpart to Solidity's `int208` operator.
     *
     * Requirements:
     *
     * - input must fit into 208 bits
     *
     * _Available since v4.7._
     */
    function toInt208(int256 value) internal pure returns (int208 downcasted) {
        downcasted = int208(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 208 bits");
    }

    /**
     * @dev Returns the downcasted int200 from int256, reverting on
     * overflow (when the input is less than smallest int200 or
     * greater than largest int200).
     *
     * Counterpart to Solidity's `int200` operator.
     *
     * Requirements:
     *
     * - input must fit into 200 bits
     *
     * _Available since v4.7._
     */
    function toInt200(int256 value) internal pure returns (int200 downcasted) {
        downcasted = int200(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 200 bits");
    }

    /**
     * @dev Returns the downcasted int192 from int256, reverting on
     * overflow (when the input is less than smallest int192 or
     * greater than largest int192).
     *
     * Counterpart to Solidity's `int192` operator.
     *
     * Requirements:
     *
     * - input must fit into 192 bits
     *
     * _Available since v4.7._
     */
    function toInt192(int256 value) internal pure returns (int192 downcasted) {
        downcasted = int192(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 192 bits");
    }

    /**
     * @dev Returns the downcasted int184 from int256, reverting on
     * overflow (when the input is less than smallest int184 or
     * greater than largest int184).
     *
     * Counterpart to Solidity's `int184` operator.
     *
     * Requirements:
     *
     * - input must fit into 184 bits
     *
     * _Available since v4.7._
     */
    function toInt184(int256 value) internal pure returns (int184 downcasted) {
        downcasted = int184(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 184 bits");
    }

    /**
     * @dev Returns the downcasted int176 from int256, reverting on
     * overflow (when the input is less than smallest int176 or
     * greater than largest int176).
     *
     * Counterpart to Solidity's `int176` operator.
     *
     * Requirements:
     *
     * - input must fit into 176 bits
     *
     * _Available since v4.7._
     */
    function toInt176(int256 value) internal pure returns (int176 downcasted) {
        downcasted = int176(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 176 bits");
    }

    /**
     * @dev Returns the downcasted int168 from int256, reverting on
     * overflow (when the input is less than smallest int168 or
     * greater than largest int168).
     *
     * Counterpart to Solidity's `int168` operator.
     *
     * Requirements:
     *
     * - input must fit into 168 bits
     *
     * _Available since v4.7._
     */
    function toInt168(int256 value) internal pure returns (int168 downcasted) {
        downcasted = int168(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 168 bits");
    }

    /**
     * @dev Returns the downcasted int160 from int256, reverting on
     * overflow (when the input is less than smallest int160 or
     * greater than largest int160).
     *
     * Counterpart to Solidity's `int160` operator.
     *
     * Requirements:
     *
     * - input must fit into 160 bits
     *
     * _Available since v4.7._
     */
    function toInt160(int256 value) internal pure returns (int160 downcasted) {
        downcasted = int160(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 160 bits");
    }

    /**
     * @dev Returns the downcasted int152 from int256, reverting on
     * overflow (when the input is less than smallest int152 or
     * greater than largest int152).
     *
     * Counterpart to Solidity's `int152` operator.
     *
     * Requirements:
     *
     * - input must fit into 152 bits
     *
     * _Available since v4.7._
     */
    function toInt152(int256 value) internal pure returns (int152 downcasted) {
        downcasted = int152(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 152 bits");
    }

    /**
     * @dev Returns the downcasted int144 from int256, reverting on
     * overflow (when the input is less than smallest int144 or
     * greater than largest int144).
     *
     * Counterpart to Solidity's `int144` operator.
     *
     * Requirements:
     *
     * - input must fit into 144 bits
     *
     * _Available since v4.7._
     */
    function toInt144(int256 value) internal pure returns (int144 downcasted) {
        downcasted = int144(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 144 bits");
    }

    /**
     * @dev Returns the downcasted int136 from int256, reverting on
     * overflow (when the input is less than smallest int136 or
     * greater than largest int136).
     *
     * Counterpart to Solidity's `int136` operator.
     *
     * Requirements:
     *
     * - input must fit into 136 bits
     *
     * _Available since v4.7._
     */
    function toInt136(int256 value) internal pure returns (int136 downcasted) {
        downcasted = int136(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 136 bits");
    }

    /**
     * @dev Returns the downcasted int128 from int256, reverting on
     * overflow (when the input is less than smallest int128 or
     * greater than largest int128).
     *
     * Counterpart to Solidity's `int128` operator.
     *
     * Requirements:
     *
     * - input must fit into 128 bits
     *
     * _Available since v3.1._
     */
    function toInt128(int256 value) internal pure returns (int128 downcasted) {
        downcasted = int128(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 128 bits");
    }

    /**
     * @dev Returns the downcasted int120 from int256, reverting on
     * overflow (when the input is less than smallest int120 or
     * greater than largest int120).
     *
     * Counterpart to Solidity's `int120` operator.
     *
     * Requirements:
     *
     * - input must fit into 120 bits
     *
     * _Available since v4.7._
     */
    function toInt120(int256 value) internal pure returns (int120 downcasted) {
        downcasted = int120(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 120 bits");
    }

    /**
     * @dev Returns the downcasted int112 from int256, reverting on
     * overflow (when the input is less than smallest int112 or
     * greater than largest int112).
     *
     * Counterpart to Solidity's `int112` operator.
     *
     * Requirements:
     *
     * - input must fit into 112 bits
     *
     * _Available since v4.7._
     */
    function toInt112(int256 value) internal pure returns (int112 downcasted) {
        downcasted = int112(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 112 bits");
    }

    /**
     * @dev Returns the downcasted int104 from int256, reverting on
     * overflow (when the input is less than smallest int104 or
     * greater than largest int104).
     *
     * Counterpart to Solidity's `int104` operator.
     *
     * Requirements:
     *
     * - input must fit into 104 bits
     *
     * _Available since v4.7._
     */
    function toInt104(int256 value) internal pure returns (int104 downcasted) {
        downcasted = int104(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 104 bits");
    }

    /**
     * @dev Returns the downcasted int96 from int256, reverting on
     * overflow (when the input is less than smallest int96 or
     * greater than largest int96).
     *
     * Counterpart to Solidity's `int96` operator.
     *
     * Requirements:
     *
     * - input must fit into 96 bits
     *
     * _Available since v4.7._
     */
    function toInt96(int256 value) internal pure returns (int96 downcasted) {
        downcasted = int96(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 96 bits");
    }

    /**
     * @dev Returns the downcasted int88 from int256, reverting on
     * overflow (when the input is less than smallest int88 or
     * greater than largest int88).
     *
     * Counterpart to Solidity's `int88` operator.
     *
     * Requirements:
     *
     * - input must fit into 88 bits
     *
     * _Available since v4.7._
     */
    function toInt88(int256 value) internal pure returns (int88 downcasted) {
        downcasted = int88(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 88 bits");
    }

    /**
     * @dev Returns the downcasted int80 from int256, reverting on
     * overflow (when the input is less than smallest int80 or
     * greater than largest int80).
     *
     * Counterpart to Solidity's `int80` operator.
     *
     * Requirements:
     *
     * - input must fit into 80 bits
     *
     * _Available since v4.7._
     */
    function toInt80(int256 value) internal pure returns (int80 downcasted) {
        downcasted = int80(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 80 bits");
    }

    /**
     * @dev Returns the downcasted int72 from int256, reverting on
     * overflow (when the input is less than smallest int72 or
     * greater than largest int72).
     *
     * Counterpart to Solidity's `int72` operator.
     *
     * Requirements:
     *
     * - input must fit into 72 bits
     *
     * _Available since v4.7._
     */
    function toInt72(int256 value) internal pure returns (int72 downcasted) {
        downcasted = int72(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 72 bits");
    }

    /**
     * @dev Returns the downcasted int64 from int256, reverting on
     * overflow (when the input is less than smallest int64 or
     * greater than largest int64).
     *
     * Counterpart to Solidity's `int64` operator.
     *
     * Requirements:
     *
     * - input must fit into 64 bits
     *
     * _Available since v3.1._
     */
    function toInt64(int256 value) internal pure returns (int64 downcasted) {
        downcasted = int64(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 64 bits");
    }

    /**
     * @dev Returns the downcasted int56 from int256, reverting on
     * overflow (when the input is less than smallest int56 or
     * greater than largest int56).
     *
     * Counterpart to Solidity's `int56` operator.
     *
     * Requirements:
     *
     * - input must fit into 56 bits
     *
     * _Available since v4.7._
     */
    function toInt56(int256 value) internal pure returns (int56 downcasted) {
        downcasted = int56(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 56 bits");
    }

    /**
     * @dev Returns the downcasted int48 from int256, reverting on
     * overflow (when the input is less than smallest int48 or
     * greater than largest int48).
     *
     * Counterpart to Solidity's `int48` operator.
     *
     * Requirements:
     *
     * - input must fit into 48 bits
     *
     * _Available since v4.7._
     */
    function toInt48(int256 value) internal pure returns (int48 downcasted) {
        downcasted = int48(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 48 bits");
    }

    /**
     * @dev Returns the downcasted int40 from int256, reverting on
     * overflow (when the input is less than smallest int40 or
     * greater than largest int40).
     *
     * Counterpart to Solidity's `int40` operator.
     *
     * Requirements:
     *
     * - input must fit into 40 bits
     *
     * _Available since v4.7._
     */
    function toInt40(int256 value) internal pure returns (int40 downcasted) {
        downcasted = int40(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 40 bits");
    }

    /**
     * @dev Returns the downcasted int32 from int256, reverting on
     * overflow (when the input is less than smallest int32 or
     * greater than largest int32).
     *
     * Counterpart to Solidity's `int32` operator.
     *
     * Requirements:
     *
     * - input must fit into 32 bits
     *
     * _Available since v3.1._
     */
    function toInt32(int256 value) internal pure returns (int32 downcasted) {
        downcasted = int32(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 32 bits");
    }

    /**
     * @dev Returns the downcasted int24 from int256, reverting on
     * overflow (when the input is less than smallest int24 or
     * greater than largest int24).
     *
     * Counterpart to Solidity's `int24` operator.
     *
     * Requirements:
     *
     * - input must fit into 24 bits
     *
     * _Available since v4.7._
     */
    function toInt24(int256 value) internal pure returns (int24 downcasted) {
        downcasted = int24(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 24 bits");
    }

    /**
     * @dev Returns the downcasted int16 from int256, reverting on
     * overflow (when the input is less than smallest int16 or
     * greater than largest int16).
     *
     * Counterpart to Solidity's `int16` operator.
     *
     * Requirements:
     *
     * - input must fit into 16 bits
     *
     * _Available since v3.1._
     */
    function toInt16(int256 value) internal pure returns (int16 downcasted) {
        downcasted = int16(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 16 bits");
    }

    /**
     * @dev Returns the downcasted int8 from int256, reverting on
     * overflow (when the input is less than smallest int8 or
     * greater than largest int8).
     *
     * Counterpart to Solidity's `int8` operator.
     *
     * Requirements:
     *
     * - input must fit into 8 bits
     *
     * _Available since v3.1._
     */
    function toInt8(int256 value) internal pure returns (int8 downcasted) {
        downcasted = int8(value);
        require(downcasted == value, "SafeCast: value doesn't fit in 8 bits");
    }

    /**
     * @dev Converts an unsigned uint256 into a signed int256.
     *
     * Requirements:
     *
     * - input must be less than or equal to maxInt256.
     *
     * _Available since v3.0._
     */
    function toInt256(uint256 value) internal pure returns (int256) {
        // Note: Unsafe cast below is okay because `type(int256).max` is guaranteed to be positive
        require(value <= uint256(type(int256).max), "SafeCast: value doesn't fit in an int256");
        return int256(value);
    }
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: @openzeppelin/contracts/utils/math/SignedMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/SignedMath.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard signed math utilities missing in the Solidity language.
 */
library SignedMath {
    /**
     * @dev Returns the largest of two signed numbers.
     */
    function max(int256 a, int256 b) internal pure returns (int256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two signed numbers.
     */
    function min(int256 a, int256 b) internal pure returns (int256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two signed numbers without overflow.
     * The result is rounded towards zero.
     */
    function average(int256 a, int256 b) internal pure returns (int256) {
        // Formula from the book "Hacker's Delight"
        int256 x = (a & b) + ((a ^ b) >> 1);
        return x + (int256(uint256(x) >> 255) & (a ^ b));
    }

    /**
     * @dev Returns the absolute unsigned value of a signed value.
     */
    function abs(int256 n) internal pure returns (uint256) {
        unchecked {
            // must be unchecked in order to support `n = type(int256).min`
            return uint256(n >= 0 ? n : -n);
        }
    }
}

// File: @openzeppelin/contracts/utils/structs/BitMaps.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/structs/BitMaps.sol)
pragma solidity ^0.8.0;

/**
 * @dev Library for managing uint256 to bool mapping in a compact and efficient way, providing the keys are sequential.
 * Largely inspired by Uniswap's https://github.com/Uniswap/merkle-distributor/blob/master/contracts/MerkleDistributor.sol[merkle-distributor].
 */
library BitMaps {
    struct BitMap {
        mapping(uint256 => uint256) _data;
    }

    /**
     * @dev Returns whether the bit at `index` is set.
     */
    function get(BitMap storage bitmap, uint256 index) internal view returns (bool) {
        uint256 bucket = index >> 8;
        uint256 mask = 1 << (index & 0xff);
        return bitmap._data[bucket] & mask != 0;
    }

    /**
     * @dev Sets the bit at `index` to the boolean `value`.
     */
    function setTo(BitMap storage bitmap, uint256 index, bool value) internal {
        if (value) {
            set(bitmap, index);
        } else {
            unset(bitmap, index);
        }
    }

    /**
     * @dev Sets the bit at `index`.
     */
    function set(BitMap storage bitmap, uint256 index) internal {
        uint256 bucket = index >> 8;
        uint256 mask = 1 << (index & 0xff);
        bitmap._data[bucket] |= mask;
    }

    /**
     * @dev Unsets the bit at `index`.
     */
    function unset(BitMap storage bitmap, uint256 index) internal {
        uint256 bucket = index >> 8;
        uint256 mask = 1 << (index & 0xff);
        bitmap._data[bucket] &= ~mask;
    }
}

// File: contracts/implementations/DropManager.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol";
import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "@openzeppelin/contracts/utils/math/SafeCast.sol";
import "@openzeppelin/contracts/utils/structs/BitMaps.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "../interfaces/IDropManager.sol";
import "../interfaces/IRedemptionManager.sol";
import "../interfaces/IRoyaltyDistributor.sol";
import "../libraries/GrtLibrary.sol";
import "../vendors/ExtendedBitmap.sol";
import "../interfaces/ISwap.sol";
import "./DropManagerAdmin.sol";

/// @title Drop Manager
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @custom:contributor Seb N
/// @custom:contributor Jo Rocca
/// @notice The nerve centre of the Droping system for GRT. Responsible for routing calls to relevant implementation
///         contracts for purchase and token logic
contract DropManager is IDropManager, DropManagerAdmin, IERC721Receiver {
  using SafeMath for uint256;
  using ExtendedBitmap for ExtendedBitmap.BitMap;
  using BitMaps for BitMaps.BitMap;
  using SafeERC20 for ERC20;

  uint128 public override releaseCounter = 0;
  uint128 public override listingCounter = 0;

  // releaseId => release info
  mapping(uint128 => Release) public override releases;

  // listingId => pending eth for listing
  mapping(uint128 => uint256) public override pendingEth;

  ExtendedBitmap.BitMap internal bidTokens;
  ExtendedBitmap.BitMap internal soldTokens;

  ITokenContract public immutable liquidToken;
  IRoyaltyDistributor public royaltyDistributor;
  IRedemptionManager public redemptionManager;
  ISwap public swapContract;

  uint256 public constant PERCENTAGE_PRECISION = 10 ** 2;
  address public royaltyWallet;

  constructor(
    address _superUser,
    address _liquidToken,
    address _redemptionManager,
    address _royaltyWallet,
    address _royaltyDistributor,
    address _swapContract
  ) DropManagerAdmin(_superUser) {
    GrtLibrary.checkZeroAddress(_superUser, "super user");
    GrtLibrary.checkZeroAddress(_liquidToken, "liquid token");
    GrtLibrary.checkZeroAddress(_redemptionManager, "redemption manager");
    GrtLibrary.checkZeroAddress(_royaltyWallet, "royalty wallet");
    GrtLibrary.checkZeroAddress(_royaltyDistributor, "royalty distributor");
    GrtLibrary.checkZeroAddress(_swapContract, "swap contract");

    liquidToken = ITokenContract(_liquidToken);
    redemptionManager = IRedemptionManager(_redemptionManager);
    royaltyDistributor = IRoyaltyDistributor(_royaltyDistributor);
    swapContract = ISwap(_swapContract);
    royaltyWallet = _royaltyWallet;
  }

  /// @dev Increment the global release counter
  ///      uint128 Represents 3.402824 × 10^38 - several orders of magnitude more releases than we expect to ever create,
  ///      hence this should never reasonably over-flow and can remain unchecked
  function _incrementReleaseCounter()
    internal
    returns (uint128 incrementedCount)
  {
    unchecked {
      incrementedCount = releaseCounter + 1;
      releaseCounter = incrementedCount;
    }
  }

  /// @dev Increment the global listing counter. This counter is shared across all listing contracts so that all listings have a unique index
  function _incrementListingCounter()
    internal
    returns (uint128 incrementedCount)
  {
    unchecked {
      incrementedCount = listingCounter + 1;
      listingCounter = incrementedCount;
    }
  }

  /// @dev Calls the liquid token contract to mint a sequential range of tokens and add tokens URIs to the registry
  /// @param qty The number of tokens to mint for the release
  /// @param liquidUri The liquid token URI to set for the batch
  /// @param redeemedUri The redeemed token URI to set for the batch
  /// @param producer The address of the producer who will receive funds for release sales
  /// @param producerPercentage The percentage of sales to send to producers. The remaining funds will be claimed by GRT
  function _createRelease(
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri,
    address producer,
    uint16 producerPercentage
  ) internal returns (Release memory release, uint128 releaseId) {
    GrtLibrary.checkZeroAddress(producer, "producer wallet");
    if (producerPercentage > 100 * PERCENTAGE_PRECISION) {
      revert InvalidProducerPercentage(producerPercentage);
    }

    releaseId = _incrementReleaseCounter();
    emit ReleaseCreated(releaseCounter);
    uint128 tokenCount = SafeCast.toUint128(
      liquidToken.mint(address(this), qty, liquidUri, redeemedUri)
    );
    release = Release({
      listingId: 0,
      startTokenId: (tokenCount - qty) + 1,
      endTokenId: tokenCount,
      listingType: 0,
      producerPercentage: producerPercentage,
      producer: producer
    });
  }

  /// @dev Create a new listing on one of the listing contracts based on the listing type
  /// @param release The release to create a listing for
  /// @param listing The listing details
  /// @param listingType The type of listing to be created (e.g English Drop or Buy It Now)
  /// @param releaseDate The date at which the listing is to be unlocked for redemption
  function _createListing(
    Release memory release,
    IListing.Listing memory listing,
    uint8 listingType,
    uint256 releaseDate,
    bytes calldata data
  ) internal returns (Release memory _release) {
    if (release.endTokenId == 0) {
      revert InvalidRelease(msg.sender, listing.releaseId);
    }
    _release = release;
    uint128 currentId = _incrementListingCounter();
    IListing listingContract = listingRegistry[listingType];

    _release.listingId = currentId;
    _release.listingType = listingType;
    listingContract.createListing(currentId, listing, data);
    if (releaseDate != 0) {
      redemptionManager.setTimeLock(listing.releaseId, releaseDate);
      emit TimeLockSet(listing.releaseId, releaseDate);
    }
  }

  /// @dev Wrapper for sending usdc to an address including error handling
  /// @param destination The address to send the amount to
  /// @param amount The amount to send
  function _callSendUsdc(address destination, uint256 amount) internal {
    GrtLibrary.checkZeroAddress(destination, "destination");
    if (amount == 0) {
      revert InvalidTokenAmount();
    }
    ERC20 token = swapContract.baseToken();
    token.safeTransfer(destination, amount);
  }

  function hasBid(uint128 tokenId) public view override returns (bool status) {
    status = bidTokens.get(tokenId);
  }

  function hasSold(uint128 tokenId) public view override returns (bool status) {
    status = soldTokens.get(tokenId);
  }

  function createRelease(
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri,
    address producer,
    uint16 producerPercentage
  ) external override onlyRole(PLATFORM_ADMIN_ROLE) {
    (Release memory release, uint128 releaseId) = _createRelease(
      qty,
      liquidUri,
      redeemedUri,
      producer,
      producerPercentage
    );
    releases[releaseId] = release;
  }

  function createListing(
    IListing.Listing calldata listing,
    uint8 listingType,
    uint256 releaseDate,
    bytes calldata data
  ) external override onlyRole(PLATFORM_ADMIN_ROLE) {
    Release memory release = releases[listing.releaseId];
    if (release.listingId != 0) {
      revert ReleaseAlreadyListed(msg.sender, listing.releaseId);
    }
    releases[listing.releaseId] = _createListing(
      release,
      listing,
      listingType,
      releaseDate,
      data
    );
  }

  function createReleaseAndList(
    uint8 listingType,
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri,
    address producer,
    uint16 producerPercentage,
    IListing.Listing memory listing,
    uint256 releaseDate,
    bytes calldata data
  ) external override onlyRole(PLATFORM_ADMIN_ROLE) {
    (Release memory release, uint128 releaseId) = _createRelease(
      qty,
      liquidUri,
      redeemedUri,
      producer,
      producerPercentage
    );
    listing.releaseId = releaseId;
    releases[releaseId] = _createListing(
      release,
      listing,
      listingType,
      releaseDate,
      data
    );
  }

  function updateListing(
    uint8 listingType,
    uint128 listingId,
    IListing.Listing calldata listing,
    bytes calldata data
  ) external override onlyRole(PLATFORM_ADMIN_ROLE) {
    IListing listingContract = listingRegistry[listingType];
    listingContract.updateListing(listingId, listing, data);
  }

  function deleteListing(
    uint8 listingType,
    uint128 listingId
  ) external onlyRole(PLATFORM_ADMIN_ROLE) {
    IListing listingContract = listingRegistry[listingType];
    listingContract.deleteListing(listingId);
  }

  function relistRelease(
    uint128 releaseId,
    uint8 newListingType,
    IListing.Listing calldata listing,
    bytes calldata data
  ) external override onlyRole(PLATFORM_ADMIN_ROLE) {
    Release memory release = releases[releaseId];
    IListing listingContract = listingRegistry[release.listingType];
    if (!(listingContract.listingEnded(release.listingId))) {
      revert ListingActive(release.listingId);
    }

    for (
      uint256 i = release.startTokenId >> 8;
      i <= release.endTokenId >> 8;
      i++
    ) {
      uint256 bid = bidTokens.getMaskedBucket(
        release.startTokenId,
        release.endTokenId,
        i * 256
      );
      uint256 sold = soldTokens.getMaskedBucket(
        release.startTokenId,
        release.endTokenId,
        i * 256
      );
      soldTokens.setBucket(i, bid | sold);
    }
    releases[releaseId] = _createListing(
      release,
      listing,
      newListingType,
      0,
      data
    );
  }

  function placeBidWithUSDC(
    uint128 releaseId,
    uint128 tokenId,
    uint256 amount,
    bytes calldata data
  ) external {
    Release memory release = releases[releaseId];
    if (allListingsPaused || pausedListings.get(release.listingId)) {
      revert ListingPaused();
    }
    _tokenInRelease(release, tokenId);
    if (soldTokens.get(tokenId)) {
      revert TokenAlreadySold();
    }
    IListing listingContract = listingRegistry[release.listingType];
    IListing.Bid memory bid = IListing.Bid({
      bidder: msg.sender,
      amount: amount
    });
    bidTokens.set(tokenId);
    ERC20 token = swapContract.baseToken();
    pendingEth[release.listingId] += amount;
    token.safeTransferFrom(msg.sender, address(this), amount);
    listingContract.registerBid(release.listingId, tokenId, bid, data);
  }

  function placeBidWithETH(
    uint128 releaseId,
    uint128 tokenId,
    uint256 amount,
    address spender,
    address payable swapTarget,
    bytes calldata data
  ) external payable {
    ERC20 token = swapContract.baseToken();
    uint256 initialBalance = token.balanceOf(address(this));
    Release memory release = releases[releaseId];
    if (allListingsPaused || pausedListings.get(release.listingId)) {
      revert ListingPaused();
    }
    _tokenInRelease(release, tokenId);
    if (soldTokens.get(tokenId)) {
      revert TokenAlreadySold();
    }
    IListing listingContract = listingRegistry[release.listingType];
    swapContract.depositETH{value: msg.value}(
      token,
      msg.value,
      address(this),
      spender,
      swapTarget,
      data
    );
    uint256 newBalance = token.balanceOf(address(this));
    if (newBalance < initialBalance.add(amount)) {
      revert IncorrectSwapParams();
    }
    uint256 _amountReceived = newBalance.sub(initialBalance);
    IListing.Bid memory bid = IListing.Bid({
      bidder: msg.sender,
      amount: _amountReceived
    });
    bidTokens.set(tokenId);
    pendingEth[release.listingId] += _amountReceived;
    listingContract.registerBid(release.listingId, tokenId, bid, data);
  }

  function placeETHBidWithReceiver(
    uint128 releaseId,
    uint128 tokenId,
    address receiver,
    uint256 amount,
    address spender,
    address payable swapTarget,
    bytes calldata data
  ) external payable {
    ERC20 token = swapContract.baseToken();
    uint256 initialBalance = token.balanceOf(address(this));
    Release memory release = releases[releaseId];
    if (allListingsPaused || pausedListings.get(release.listingId)) {
      revert ListingPaused();
    }
    _tokenInRelease(release, tokenId);
    if (soldTokens.get(tokenId)) {
      revert TokenAlreadySold();
    }
    IListing listingContract = listingRegistry[release.listingType];
    swapContract.depositETH{value: msg.value}(
      token,
      msg.value,
      address(this),
      spender,
      swapTarget,
      data
    );
    uint256 _amountReceived = token.balanceOf(address(this)).sub(initialBalance);
    if (_amountReceived < amount) {
      revert IncorrectSwapParams();
    }
    IListing.Bid memory bid = IListing.Bid({
      bidder: receiver,
      amount: _amountReceived
    });
    bidTokens.set(tokenId);
    pendingEth[release.listingId] += _amountReceived;
    listingContract.registerBid(release.listingId, tokenId, bid, data);
  }

  function transferBaseToken(
    uint8 listingType,
    uint128 listingId,
    address destination,
    uint256 amount
  ) external override onlyListingOperator(listingType) {
    pendingEth[listingId] -= amount;
    _callSendUsdc(destination, amount);
  }

  function transferToken(
    uint8 listingType,
    uint128 tokenId,
    address destination
  ) external onlyListingOperator(listingType) {
    soldTokens.set(tokenId);
    liquidToken.safeTransferFrom(address(this), destination, tokenId);
  }

  function claimToken(
    uint128 releaseId,
    uint128 tokenId,
    uint128 listingId,
    uint8 listingType
  ) external override {
    Release memory release = releases[releaseId];
    _tokenInRelease(release, tokenId);
    IListing listingInstance = listingRegistry[listingType];
    emit TokenClaimed(msg.sender, tokenId);
    uint256 saleAmount = listingInstance.validateTokenClaim(
      listingId,
      releaseId,
      tokenId,
      msg.sender
    );
    if (saleAmount > 0) {
      _distributeFunds(
        releaseId,
        listingId,
        release.producer,
        release.producerPercentage,
        saleAmount
      );
    }
    liquidToken.safeTransferFrom(address(this), msg.sender, tokenId);
  }

  function withdrawTokens(
    uint128 releaseId,
    uint128[] calldata tokens,
    address destination
  ) external override onlyRole(PLATFORM_ADMIN_ROLE) {
    Release memory release = releases[releaseId];
    IListing listingInstance = listingRegistry[release.listingType];
    if (!(listingInstance.listingEnded(release.listingId))) {
      revert ListingActive(release.listingId);
    }
    for (uint16 i = 0; i < tokens.length; i++) {
      _tokenInRelease(release, tokens[i]);
      if (bidTokens.get(tokens[i]) || soldTokens.get(tokens[i])) {
        revert TokenHasBid(tokens[i]);
      }
      soldTokens.set(tokens[i]);
      liquidToken.safeTransferFrom(address(this), destination, tokens[i]);
    }
    emit TokensWithdrawn(destination, release.listingId, tokens);
  }

  function distributeListingFunds(
    uint8 listingType,
    uint128 listingId,
    uint128 releaseId
  ) external onlyRole(PLATFORM_ADMIN_ROLE) {
    Release memory release = releases[releaseId];
    IListing listingInstance = listingRegistry[listingType];

    if (!listingInstance.validateManualDistribution(listingId)) {
      revert IListing.DistributionNotSupported();
    }

    uint256 totalRemainingSales = pendingEth[listingId];

    if (totalRemainingSales == 0) {
      revert NoFundsRemaining(listingId);
    }

    _distributeFunds(
      releaseId,
      release.listingId,
      release.producer,
      release.producerPercentage,
      totalRemainingSales
    );
  }

  function distributeSaleFunds(
    uint8 listingType,
    uint128 listingId,
    uint128 releaseId,
    uint256 saleAmount
  ) external onlyListingOperator(listingType) {
    Release memory release = releases[releaseId];
    _distributeFunds(
      releaseId,
      listingId,
      release.producer,
      release.producerPercentage,
      saleAmount
    );
  }

  function distributeSecondaryFunds(
    uint128 releaseId,
    uint256 amount
  ) external onlyRole(PLATFORM_ADMIN_ROLE) {
    Release memory release = releases[releaseId];
    royaltyDistributor.distributeFunds(
      amount,
      release.producer,
      release.producerPercentage,
      royaltyWallet,
      releaseId
    );
  }

  /// @dev Distribute funds between a receiver and the GRT royalty wallet. Any remaining funds (100% - receiver percentage) will be sent to the royalty wallet
  /// @param releaseId The id of the release for which the listing was made
  /// @param listingId The id of the listing for the distibution to decrement the pending sale amount
  /// @param receiver The receiver of the percentage royalties (i.e a producer)
  /// @param receiverPercentage The percentage of the sale amount to send to the receiver
  /// @param saleAmount The value of the sale to distribute in wei
  function _distributeFunds(
    uint128 releaseId,
    uint128 listingId,
    address receiver,
    uint256 receiverPercentage,
    uint256 saleAmount
  ) internal {
    GrtLibrary.checkZeroAddress(receiver, "receiver address");
    uint256 receiverAmount = saleAmount.mul(receiverPercentage).div(
      PERCENTAGE_PRECISION * 100
    );
    uint256 royaltyAmount = saleAmount.sub(receiverAmount);
    pendingEth[listingId] -= saleAmount;

    emit FundsDistributed(receiver, receiverAmount, releaseId);
    emit FundsDistributed(royaltyWallet, royaltyAmount, releaseId);

    _callSendUsdc(receiver, receiverAmount);
    _callSendUsdc(royaltyWallet, royaltyAmount);
  }

  /// @dev Check that a token exists within the bounds of a release (between the start and end token id, inclusive)
  /// @param release The release to check
  /// @param tokenId The id of the token to check
  function _tokenInRelease(
    Release memory release,
    uint128 tokenId
  ) internal pure {
    if (release.startTokenId > tokenId || tokenId > release.endTokenId) {
      revert TokenNotInRelease(tokenId);
    }
  }

  function onERC721Received(
    address,
    address,
    uint256,
    bytes calldata
  ) external pure override returns (bytes4) {
    return IERC721Receiver.onERC721Received.selector;
  }

  //#########################
  //####### SETTERS ########

  function setRedemptionManager(
    address _redemptionManager
  ) external onlyRole(PLATFORM_ADMIN_ROLE) {
    GrtLibrary.checkZeroAddress(_redemptionManager, "redemption manager");
    redemptionManager = IRedemptionManager(_redemptionManager);
  }

  function setRoyaltyDistributor(
    address _royaltyDistributor
  ) external onlyRole(PLATFORM_ADMIN_ROLE) {
    GrtLibrary.checkZeroAddress(_royaltyDistributor, "royalty distributor");
    royaltyDistributor = IRoyaltyDistributor(_royaltyDistributor);
  }

  function setRoyaltyWallet(
    address _royaltyWallet
  ) external onlyRole(PLATFORM_ADMIN_ROLE) {
    GrtLibrary.checkZeroAddress(_royaltyWallet, "royalty wallet");
    royaltyWallet = _royaltyWallet;
  }

  modifier onlyListingOperator(uint8 listingType) {
    if (msg.sender != address(listingRegistry[listingType])) {
      revert InvalidTransferOperator();
    }
    _;
  }
}

// File: contracts/implementations/DropManagerAdmin.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "@openzeppelin/contracts/access/AccessControl.sol";
import "@openzeppelin/contracts/utils/structs/BitMaps.sol";
import "../interfaces/IDropManagerAdmin.sol";
import "../interfaces/IGrtWines.sol";
import "../interfaces/IListing.sol";
import "../libraries/GrtLibrary.sol";

/// @title Drop Manager Admin
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @notice Provides admin functionality to the Drop Manager for the GRT Wines platform
contract DropManagerAdmin is IDropManagerAdmin, IGrtWines, AccessControl {
  using BitMaps for BitMaps.BitMap;
  bytes32 public constant override PLATFORM_ADMIN_ROLE =
    keccak256("PLATFORM_ADMIN_ROLE");

  bool public override allListingsPaused = false;

  // Bitmap of listings that are current paused
  BitMaps.BitMap internal pausedListings;

  // Listing type ID  => listing contract instance
  mapping(uint8 => IListing) public override listingRegistry;
  // Listing contract address => Listing type ID
  mapping(address => uint8) public override addressListingLookup;

  constructor(address superUser) {
    GrtLibrary.checkZeroAddress(superUser, "super user");
    _grantRole(DEFAULT_ADMIN_ROLE, superUser);
  }

  /// @dev Set the status of all listings to {status}
  /// @param status The new status to set
  function _changeAllPause(bool status) internal {
    if (allListingsPaused == status) {
      revert ListingStatusAlreadySet();
    }
    allListingsPaused = status;
    emit AllListingStatusChanged(msg.sender, status);
  }

  function registerListingType(uint8 listingType, address listingContract)
    external
    onlyRole(PLATFORM_ADMIN_ROLE)
  {
    GrtLibrary.checkZeroAddress(listingContract, "listing contract");
    if (address(listingRegistry[listingType]) != address(0)) {
      revert ListingIdTaken();
    }
    listingRegistry[listingType] = IListing(listingContract);
    addressListingLookup[listingContract] = listingType;
    emit ListingTypeCreated(listingContract, listingType);
  }

  function pauseListing(uint128 listingId)
    external
    onlyRole(PLATFORM_ADMIN_ROLE)
  {
    if (pausedListings.get(listingId) == true) {
      revert ListingStatusAlreadySet();
    }
    pausedListings.setTo(listingId, true);
    emit ListingStatusChanged(listingId, true);
  }

  function unpauseListing(uint128 listingId)
    external
    onlyRole(PLATFORM_ADMIN_ROLE)
  {
    if (pausedListings.get(listingId) == false) {
      revert ListingStatusAlreadySet();
    }
    pausedListings.setTo(listingId, false);
    emit ListingStatusChanged(listingId, false);
  }

  function listingPauseStatus(uint128 listingId)
    external
    view
    returns (bool status)
  {
    status = pausedListings.get(listingId);
  }

  function pauseAllListings() external onlyRole(PLATFORM_ADMIN_ROLE) {
    _changeAllPause(true);
  }

  function unpauseAllListings() external onlyRole(PLATFORM_ADMIN_ROLE) {
    _changeAllPause(false);
  }
}

// File: contracts/implementations/TokenContract.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                    
*/

import "@openzeppelin/contracts/token/ERC721/extensions/ERC721Royalty.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/access/AccessControl.sol";
import "@openzeppelin/contracts/utils/math/SafeCast.sol";
import "@openzeppelin/contracts/utils/Counters.sol";
import "@openzeppelin/contracts/utils/Strings.sol";
import "operator-filter-registry/src/DefaultOperatorFilterer.sol";
import "../interfaces/IInsuranceRegistry.sol";
import "../interfaces/ITokenContract.sol";
import "../interfaces/ITokenRegistry.sol";
import "../libraries/GrtLibrary.sol";

/// @title TokenContract
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @custom:contributor Brodie S
/// @notice Implementation to be used for the Liquid and Redeemed editions of each token
/// @dev  The GRT Wines architecture uses a dual ERC721 token system. When releases are created the `LiquidToken`
///       is minted and can be purchased through various listing mechanisms via the Drop Manager. When a user
///       wishes to redeem their token for a physical asset, the `LiquidToken` is burned, and a `RedeemedToken` is
///       minted. The same `TokenContract` implementation is deployed twice, once for each edition. The metadata
///       for both the Liquid and Redeemed editions of each token is set when a release is created, and manage by
///       the `TokenRegistry`
contract TokenContract is
  DefaultOperatorFilterer,
  ITokenContract,
  AccessControl,
  ERC721Royalty
{
  using Counters for Counters.Counter;
  using Strings for uint256;

  //#########################
  //#### STATE VARIABLES ####

  Counters.Counter private _tokenIdCounter;

  bytes32 public constant override PLATFORM_ADMIN_ROLE =
    keccak256("PLATFORM_ADMIN_ROLE");
  bytes32 public constant override MINTER_ROLE = keccak256("MINTER_ROLE");
  bytes32 public constant override BURNER_ROLE = keccak256("BURNER_ROLE");

  ITokenRegistry public override tokenRegistry;
  IInsuranceRegistry public override insuranceRegistry;
  address public override redemptionManager;

  bool public immutable useRedeemedUri;

  //#########################
  //#### IMPLEMENTATION ####

  constructor(
    string memory _name,
    string memory _symbol,
    address _platformAdmin,
    address _superUser,
    address _insuranceRegistry,
    address _tokenRegistry,
    bool _useRedeemedUri,
    address _secondaryRoyaltyReceiver,
    uint96 _secondaryRoyaltyFee
  ) ERC721(_name, _symbol) {
    GrtLibrary.checkZeroAddress(_platformAdmin, "platform admin");
    GrtLibrary.checkZeroAddress(_superUser, "super user");
    GrtLibrary.checkZeroAddress(_insuranceRegistry, "insurance registry");
    GrtLibrary.checkZeroAddress(_tokenRegistry, "token registry");
    GrtLibrary.checkZeroAddress(_secondaryRoyaltyReceiver, "secondary royalty");

    insuranceRegistry = IInsuranceRegistry(_insuranceRegistry);
    tokenRegistry = ITokenRegistry(_tokenRegistry);
    _setupRole(PLATFORM_ADMIN_ROLE, _platformAdmin);
    _setupRole(DEFAULT_ADMIN_ROLE, _superUser);
    _setRoleAdmin(MINTER_ROLE, PLATFORM_ADMIN_ROLE);
    _setRoleAdmin(BURNER_ROLE, PLATFORM_ADMIN_ROLE);

    useRedeemedUri = _useRedeemedUri;
    _setDefaultRoyalty(_secondaryRoyaltyReceiver, _secondaryRoyaltyFee);
  }

  /// @dev If either of the from or to fields are 0 address, this is a mint or burn, return early to continue without storage read
  /// @dev See @openzeppelin ERC721.sol for further details
  function _beforeTokenTransfer(
    address from,
    address to,
    uint256 tokenId,
    uint256 /* batch size -- not used in standard ERC721 */
  ) internal view override {
    if (from == address(0) || to == address(0)) {
      return;
    }
    if (
      insuranceRegistry.checkTokenStatus(tokenId) &&
      msg.sender != redemptionManager
    ) {
      revert InsuranceEventRegistered(tokenId);
    }
  }

  /// @dev Overrides standard tokenURI method to retrieve the URI from the Token Registry based on the token type (liquid or redeemed)
  /// @param tokenId The id of the token to retrieve a URI for
  /// @return The token URI
  function tokenURI(uint256 tokenId)
    public
    view
    override
    returns (string memory)
  {
    _requireMinted(tokenId);

    ITokenRegistry.TokenKey memory tokenKey = tokenRegistry.getTokenKey(
      tokenId
    );

    string memory baseURI = useRedeemedUri
      ? tokenKey.redeemedUri
      : tokenKey.liquidUri;

    uint256 tokenIndex = tokenId - tokenKey.key + 1;

    return
      bytes(baseURI).length > 0
        ? string(abi.encodePacked(baseURI, tokenIndex.toString()))
        : "";
  }

  function mint(
    address receiver,
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri
  ) external override returns (uint256 mintCount) {
    bool canMint = hasRole(MINTER_ROLE, msg.sender) ||
      hasRole(PLATFORM_ADMIN_ROLE, msg.sender);
    if (!canMint) {
      revert IncorrectAccess(msg.sender);
    }
    uint256 startToken = _tokenIdCounter.current() + 1;
    for (uint16 i = 0; i < qty; i++) {
      _tokenIdCounter.increment();
      uint256 currentToken = _tokenIdCounter.current();
      _safeMint(receiver, currentToken);
    }

    tokenRegistry.addBatchMetadata(
      ITokenRegistry.TokenKey(
        liquidUri,
        redeemedUri,
        startToken,
        SafeCast.toUint16(qty),
        false
      )
    );
    return _tokenIdCounter.current();
  }

  function mintWithId(MintWithIdArgs[] calldata mintWithIdArgs)
    external
    override
  {
    if (!hasRole(MINTER_ROLE, msg.sender)) {
      revert IncorrectAccess(msg.sender);
    }
    for (uint16 i = 0; i < mintWithIdArgs.length; i++) {
      _safeMint(mintWithIdArgs[i].to, mintWithIdArgs[i].tokenId);
    }
  }

  function burn(uint256[] calldata tokens) external override {
    bool canBurn = hasRole(BURNER_ROLE, msg.sender) ||
      hasRole(PLATFORM_ADMIN_ROLE, msg.sender);
    if (!canBurn) {
      revert IncorrectAccess(msg.sender);
    }
    for (uint16 i = 0; i < tokens.length; i++) {
      _burn(tokens[i]);
    }
  }

  function changeTokenMetadata(
    uint256 batchIndex,
    string memory liquidUri,
    string memory redeemedUri
  ) external override {
    if (!hasRole(PLATFORM_ADMIN_ROLE, msg.sender)) {
      revert IncorrectAccess(msg.sender);
    }
    tokenRegistry.updateBatchMetadata(batchIndex, liquidUri, redeemedUri);
  }

  function lockTokenMetadata(uint256 batchIndex) external override {
    if (!hasRole(PLATFORM_ADMIN_ROLE, msg.sender)) {
      revert IncorrectAccess(msg.sender);
    }
    tokenRegistry.lockBatchMetadata(batchIndex);
  }

  function tokenLocked(uint256 tokenId)
    external
    view
    override
    returns (bool hasUpdated)
  {
    return tokenRegistry.getTokenKey(tokenId).locked;
  }

  function setInsuranceRegistry(address _registryAddress)
    external
    override
    onlyRole(PLATFORM_ADMIN_ROLE)
  {
    GrtLibrary.checkZeroAddress(_registryAddress, "insurance registry");
    insuranceRegistry = IInsuranceRegistry(_registryAddress);
  }

  function setRedemptionManager(address _managerAddress)
    external
    override
    onlyRole(PLATFORM_ADMIN_ROLE)
  {
    GrtLibrary.checkZeroAddress(_managerAddress, "platform manager");
    redemptionManager = _managerAddress;
  }

  function setSecondaryRoyalties(address receiver, uint96 feeNumerator)
    external
    override
    onlyRole(PLATFORM_ADMIN_ROLE)
  {
    GrtLibrary.checkZeroAddress(receiver, "secondary royalty");
    _setDefaultRoyalty(receiver, feeNumerator);
  }

  /// @dev Due to multiple inhereted Open Zeppelin contracts implementing supportsInterface we must provide an override as
  /// below so Solidity knows how to resolve conflicted inheretence
  /// see https://github.com/OpenZeppelin/openzeppelin-contracts/issues/3107
  function supportsInterface(bytes4 interfaceId)
    public
    view
    override(IERC165, ERC721Royalty, AccessControl)
    returns (bool)
  {
    return super.supportsInterface(interfaceId);
  }

  /// @dev Overrides required for the Operator Filter Registry
  /// see https://github.com/ProjectOpenSea/operator-filter-registry
  function setApprovalForAll(address operator, bool approved)
    public
    override(ERC721, IERC721)
    onlyAllowedOperatorApproval(operator)
  {
    super.setApprovalForAll(operator, approved);
  }

  function approve(address operator, uint256 tokenId)
    public
    override(ERC721, IERC721)
    onlyAllowedOperatorApproval(operator)
  {
    super.approve(operator, tokenId);
  }

  function transferFrom(
    address from,
    address to,
    uint256 tokenId
  ) public override(ERC721, IERC721) onlyAllowedOperator(from) {
    super.transferFrom(from, to, tokenId);
  }

  function safeTransferFrom(
    address from,
    address to,
    uint256 tokenId
  ) public override(ERC721, IERC721) onlyAllowedOperator(from) {
    super.safeTransferFrom(from, to, tokenId);
  }

  function safeTransferFrom(
    address from,
    address to,
    uint256 tokenId,
    bytes memory data
  ) public override(ERC721, IERC721) onlyAllowedOperator(from) {
    super.safeTransferFrom(from, to, tokenId, data);
  }
}

// File: contracts/interfaces/IDropManager.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "./IDropManagerAdmin.sol";
import "./IListing.sol";
import "./ITokenContract.sol";
import "./IRedemptionManager.sol";
import "./IRoyaltyDistributor.sol";
import "./ISwap.sol";

/// @title Drop Manager
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @custom:contributor Seb N
/// @notice The nerve centre of the Droping system for GRT. Responsible for routing calls to relevant implementation
///         contracts for purchase and token logic
interface IDropManager is IDropManagerAdmin {
  //################
  //#### STRUCTS ####

  /// @dev Parameters for creating a Release which could include one or more tokens. E.g a box of six wines
  /// @param listingType The identifier of the drop contract which manages sale logic. Up to 255
  /// @param listingId The listing identifier - Counter from DropManager contract
  /// @param startTokenId Start point of included tokens
  /// @param endTokenId The end of the token range to drop
  /// @param producerPercentage The percentage of sales to send to producers. The remaining funds will be claimed by GRT
  /// @param producer The address of the producer who will receive funds for release sales
  struct Release {
    uint128 listingId;
    uint128 startTokenId;
    uint128 endTokenId;
    uint8 listingType;
    uint16 producerPercentage;
    address producer;
  }

  //################
  //#### EVENTS ####

  /// @dev Emitted when a Release is created
  event ReleaseCreated(uint128 releaseId);

  /// @dev Emitted on successful token claim
  event TokenClaimed(address claimant, uint128 tokenId);

  /// @dev Emitted when unsold tokens are withdrawn
  event TokensWithdrawn(
    address indexed receiver,
    uint128 listingId,
    uint128[] tokens
  );

  /// @dev Emitted when proceeds from listings are withdrawn successfully
  event ProceedsWithdrawn(address indexed receiver, uint128[] listingIds);

  /// @dev Emitted on successful setting of a Time Lock
  event TimeLockSet(uint256 indexed releaseId, uint256 indexed releaseDate);

  /// @dev Emitted when funds are distributed to producers
  event FundsDistributed(address receiver, uint256 amount, uint128 releaseId);

  //################
  //#### ERRORS ####

  /// @dev Thrown if an account attempts to create a listing for a release that already has a listing
  error ReleaseAlreadyListed(address sender, uint128 releaseId);

  /// @dev Thrown if the provided releaseId does not exist
  error InvalidRelease(address sender, uint128 releaseId);

  /// @dev Thrown if an account attempts to withdraw an un-sold token that is not included in the release ID provided
  error TokenNotInRelease(uint128 tokenId);

  /// @dev Thrown if usdc balance of contract does not match expected after swap
  error IncorrectSwapParams();

  /// @dev Thrown if an account attempts to withdraw a token that already has a bid
  error TokenHasBid(uint128 tokenId);

  /// @dev Thrown everytime unless msg.sender is the address of the listingRegistry itself
  error InvalidTransferOperator();

  /// @dev Thrown if an ETH transfer fails
  error EthTransferFailed();

  /// @dev Thrown if attempting to transfer a zero eth amount
  error InvalidTokenAmount();

  /// @dev Thrown if operations attempt to be performed on a release with a listing that is still active
  /// @param listingId The listing ID that is still considered active
  error ListingActive(uint128 listingId);

  /// @dev Thrown if the token has already sold or been otherwise withdrawn
  error TokenAlreadySold();

  /// @dev Thrown if an invalid producer percentage is provided (greater than 100%)
  error InvalidProducerPercentage(uint16 producerPercentage);

  /// @dev Throw if attempting to distribute funds on a release that does not have any remaining funds to distribute
  /// @param listingId The id of the listing for which the error was thrown
  error NoFundsRemaining(uint128 listingId);

  //###################
  //#### FUNCTIONS ####

  /// @notice Create a release for tokens
  /// @dev Calls the liquid token contract to mint a sequential range of tokens and add tokens URIs to the registry
  /// @param qty The number of tokens to mint for the release
  /// @param liquidUri The liquid token URI to set for the batch
  /// @param redeemedUri The redeemed token URI to set for the batch
  /// @param producer The address of the producer who will receive funds for release sales
  /// @param producerPercentage The percentage of sales to send to producers. The remaining funds will be claimed by GRT
  function createRelease(
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri,
    address producer,
    uint16 producerPercentage
  ) external;

  /// @notice Create a listing for a release
  /// @dev Creates a Listing at the target contract based on the provided listingType
  /// @param listing The listing data
  /// @param listingType The type of listing this should be e.g EnglishDrop, Buy It Now
  /// @param releaseDate The date at which the listing is published.
  /// @param data Arbitrary additional data to be passed to the Listing contract, should additional data be required by new listing types in future
  function createListing(
    IListing.Listing calldata listing,
    uint8 listingType,
    uint256 releaseDate,
    bytes calldata data
  ) external;

  /// @notice Create a release and list in at a listing contract with one call
  /// @param listingType The listing type identifer
  /// @param qty The number of tokens to mint
  /// @param liquidUri The liquid token URI to set for the batch
  /// @param redeemedUri The redeemed token URI to set for the batch
  /// @param producer The address of the producer who will receive funds for release sales
  /// @param producerPercentage The percentage of sales to send to producers. The remaining funds will be claimed by GRT
  /// @param listing The listing data to pass to the listing contract
  /// @param releaseDate The date at which the release is published.
  /// @param data Arbitrary additional data should requirements change in future
  function createReleaseAndList(
    uint8 listingType,
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri,
    address producer,
    uint16 producerPercentage,
    IListing.Listing memory listing,
    uint256 releaseDate,
    bytes calldata data
  ) external;

  /// @notice Update a listing
  /// @dev Cannot update an active listing
  /// @param listingType The listing type identifer
  /// @param listingId The identifier of the listing to be updated
  /// @param listing The listing data to update the existing listing with
  /// @param data Arbitrary additional data should requirements change in future
  function updateListing(
    uint8 listingType,
    uint128 listingId,
    IListing.Listing calldata listing,
    bytes calldata data
  ) external;

  /// @notice Delete a listing
  /// @dev Cannot delete a listing once it has started
  /// @param listingType The listing type ID of the target listing contract
  /// @param listingId The listing ID of the listing itself
  function deleteListing(uint8 listingType, uint128 listingId) external;

  /// @notice Relist a release, maintaining funds stored in the DropManager
  /// @dev Assigns a new listing ID and sets new listing information on the target contract
  /// @dev Calls create listing on the target contract, even if the target is the same as the old one
  /// @param releaseId The release ID this relisting targets
  /// @param listing The listing information to relist with
  /// @param data Arbitrary additional data
  function relistRelease(
    uint128 releaseId,
    uint8 newListingType,
    IListing.Listing calldata listing,
    bytes calldata data
  ) external;

  /// @notice Place a bid directly with USDC
  /// @param releaseId The release ID this relisting targets
  /// @param tokenId The targeted token ID
  /// @param amount The amount of USDC for the token
  /// @param data Arbitrary additional data
  function placeBidWithUSDC(
    uint128 releaseId,
    uint128 tokenId,
    uint256 amount,
    bytes calldata data
  ) external;

  /// @notice Places a bid with ETH
  /// @dev Calls Swap contract to exchange ETH for USDC
  /// @param releaseId The release ID this relisting targets
  /// @param tokenId The targeted token ID
  /// @param spender an address provided by the 0x Quote API
  /// @param swapTarget an address provided by the 0x Quote API
  /// @param data Data from 0x Swap API
  function placeBidWithETH(
    uint128 releaseId,
    uint128 tokenId,
    uint256 amount,
    address spender,
    address payable swapTarget,
    bytes calldata data
  ) external payable;

  /// @notice Places a bid with FIAT through integration iwth Paper
  /// @dev Calls Swap contract to exchange ETH for USDC
  /// @param releaseId The release ID this relisting targets
  /// @param tokenId The targeted token ID
  /// @param receiver The person to receive the token through purchase with Paper
  /// @param spender an address provided by the 0x Quote API
  /// @param swapTarget an address provided by the 0x Quote API
  /// @param data Data from 0x Swap API
  function placeETHBidWithReceiver(
    uint128 releaseId,
    uint128 tokenId,
    address receiver,
    uint256 amount,
    address spender,
    address payable swapTarget,
    bytes calldata data
  ) external payable;

  /// @notice Callback function for listing contracts to return USDC of the previous highest bidder once they are out-bid
  /// @dev Can ONLY be used as a callback from a listing contract - if msg.sender != listingRegistry[listingId] it reverts
  /// @param listingType The listing ID of the contract doing the callback
  /// @param listingId The ID of the listing
  /// @param destination The destination for the USDC transfer (previous higest bidder)
  /// @param amount The value of ETH to send
  function transferBaseToken(
    uint8 listingType,
    uint128 listingId,
    address destination,
    uint256 amount
  ) external;

  /// @notice Callback function for listing contracts to send tokens to a user for immediate settlement listings - e.g buy it now
  /// @dev Can ONLY be used as a callback from a listing contract - if msg.sender != listingRegistry[listingId] it reverts
  /// @param listingType The listing ID of the contract doing the callback
  /// @param tokenId The ID of the token to be transferred
  /// @param destination The destination for the ETH transfer (previous higest bidder)
  function transferToken(
    uint8 listingType,
    uint128 tokenId,
    address destination
  ) external;

  /// @notice Claim a token that has been won via a non-immediate settlement sale, i.e Drop
  /// @dev Performs check at the target contract to verify the highest bidder, listing has ended etc
  /// @param tokenId The ID of the token to be checked
  /// @param listingId The ID of the listing this token was won in
  /// @param listingType The listing type this was one from
  function claimToken(
    uint128 releaseId,
    uint128 tokenId,
    uint128 listingId,
    uint8 listingType
  ) external;

  /// @notice Withdraw un-sold tokens.
  /// @dev If tokens are withdrawn from the dropManager they cannot be relisted for sale via the current droping system
  /// @param releaseId The release ID these tokens belong to
  /// @param tokens Array of token IDs
  /// @param destination The destination for the tokens to be withdrawn to
  function withdrawTokens(
    uint128 releaseId,
    uint128[] calldata tokens,
    address destination
  ) external;

  /// @notice Distribute remaining funds for a release that usually distributes funds on claim (e.g Drops)
  /// @dev Distributes to the assigned producer wallet, with remaining funds going to the grt royalty wallet
  /// @dev Should only be called for releases that store pending eth balances until claim (e.g Drops)
  /// @dev Emits a {FundsDistributed} event for the receiver
  /// @dev Emits a {FundsDistributed} event for the royalties
  /// @dev Can only be called by PLATFORM_ADMIN role
  /// @param listingType The id of the listing contract in which the listing exists
  /// @param listingId The id of the listing to release funds for
  /// @param releaseId The id of the release for which the listing was made
  function distributeListingFunds(
    uint8 listingType,
    uint128 listingId,
    uint128 releaseId
  ) external;

  /// @notice Distribute funds for a single sale to producer account, with remaining funds sent as a royalty to the grt royalty account
  /// @dev Can only be called by Listing contracts
  /// @dev Should be called when the token is claimed for drops, and on sale for instant transactions like Buy It Now
  /// @dev Emits a {FundsDistributed} event for the receiver
  /// @dev Emits a {FundsDistributed} event for the royalties
  /// @param listingType The type that this call was made from
  /// @param listingId The ID of the listing the token was won in
  /// @param releaseId The ID of the release the token was won in
  /// @param saleAmount The total value of the sale, with producer proceed being a percentage of this amount
  function distributeSaleFunds(
    uint8 listingType,
    uint128 listingId,
    uint128 releaseId,
    uint256 saleAmount
  ) external;

  /// @notice Distribute secondary market sales for a release
  /// @dev Distributes to the assigned producer wallet, with remaining funds going to the grt royalty wallet
  /// @dev Emits a {FundsDistributed} event for the receiver
  /// @dev Emits a {FundsDistributed} event for the royalties
  /// @dev Can only be called by PLATFORM_ADMIN role
  /// @param releaseId The id of the release for which the listing was made
  /// @param amount The amount of funds to distribute
  function distributeSecondaryFunds(uint128 releaseId, uint256 amount) external;

  /// @notice Getter for specific bit in hasBid bitmap
  /// @param tokenId The token ID
  /// @return status Whether or not the token has a bid
  function hasBid(uint128 tokenId) external view returns (bool status);

  /// @notice Getter for specific bit in hasSold bitmap
  /// @param tokenId The token ID
  /// @return  status Whether or not the token has sold
  function hasSold(uint128 tokenId) external view returns (bool status);

  //################################
  //#### AUTO-GENERATED GETTERS ####
  function releaseCounter() external returns (uint128 currentValue);

  function listingCounter() external returns (uint128 currentValue);

  /// @notice Setter for setting the redemption manager.
  /// @dev sets the address for the redemption manager so that calls to the Redemption Manager can be made.
  /// @param _redemptionManager the address of the redemption manager.
  function setRedemptionManager(address _redemptionManager) external;

  /// @notice Setter for setting the royalty distributor.
  /// @dev sets the address for the redemption manager so that calls to the Redemption Manager can be made.
  /// @param _royaltyDistributor the address of the royalty distributor.
  function setRoyaltyDistributor(address _royaltyDistributor) external;

  /// @notice Setter for the GRT Royalty Walet address
  /// @dev Sets the address for the wallet that receives royalties from token sales
  /// @param _royaltyWallet the address of the royalty wallet
  function setRoyaltyWallet(address _royaltyWallet) external;

  function releases(
    uint128 releaseId
  )
    external
    returns (
      uint128 listingId,
      uint128 startTokenId,
      uint128 endTokenId,
      uint8 listingType,
      uint16 producerPercentage,
      address producer
    );

  function pendingEth(uint128 listingId) external returns (uint256 pending);

  function liquidToken() external returns (ITokenContract tokenContract);

  function royaltyDistributor()
    external
    returns (IRoyaltyDistributor _royaltyDistributor);

  function redemptionManager()
    external
    returns (IRedemptionManager _redemptionManager);

  function swapContract() external returns (ISwap _swap);
}

// File: contracts/interfaces/IDropManagerAdmin.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "./IListing.sol";

/// @title Drop Manager Admin
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @notice Provides admin functionality to the Drop Manager for the GRT Wines platform
interface IDropManagerAdmin {
  //################
  //#### EVENTS ####

  /// @dev Emitted when a listing type is created
  /// @param listingAddress The address of the listing contract
  /// @param listingType The ID to assign to the listing
  event ListingTypeCreated(
    address indexed listingAddress,
    uint8 indexed listingType
  );

  /// @dev Emitted when the status of a listing is paused / unpaused
  /// @param listingType The ID of the listing that had it's status changed
  /// @param status The status the listing was changed to
  event ListingStatusChanged(uint128 indexed listingType, bool status);

  /// @dev Emitted when all listings are globally paused / unpaused
  /// @param sender The sender of the transaction
  /// @param status The status that the global pause / unpause was changed to
  event AllListingStatusChanged(address indexed sender, bool status);

  //################
  //#### ERRORS ####

  /// @dev Thrown if the specific listing being accessed, or all listings are paused
  error ListingPaused();
  /// @dev Thrown if the listing ID has been taken by an existing implementaiton
  error ListingIdTaken();
  /// @dev Thrown if changing the pause status is a redundant call
  error ListingStatusAlreadySet();

  //###################
  //#### FUNCTIONS ####

  /// @notice Use to register a listing type logic contract
  /// @dev IDs are not sequential and it is assumed that the sender of this transaction has some intelligence around how they use this
  /// @param listingType The ID of the listing type to be created
  /// @param listingContract The address of the listing contract
  function registerListingType(uint8 listingType, address listingContract)
    external;

  /// @notice Pause a specific listing
  /// @dev Specific pause function so that this operation is idempotent
  /// @param listingId The ID of the listing to pause
  function pauseListing(uint128 listingId) external;

  /// @notice Unpause a specific listing
  /// @dev Specific unpause function so that this operation is idempotent
  /// @param listingId The ID of the listing to unpause
  function unpauseListing(uint128 listingId) external;

  /// @notice Pause all listings
  /// @dev Specific pause function so that this operation is idempotent
  function pauseAllListings() external;

  /// @notice Unpause all listings
  /// @dev Specific unpause function so that this operation is idempotent
  function unpauseAllListings() external;

  //################################
  //#### AUTO-GENERATED GETTERS ####

  function allListingsPaused() external returns (bool);

  function listingRegistry(uint8) external returns (IListing);

  function addressListingLookup(address) external returns (uint8);

  function PLATFORM_ADMIN_ROLE() external returns (bytes32);
}

// File: contracts/interfaces/IGrtWines.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

interface IGrtWines {
  /// @dev Thrown if the sender has incorrect access to use a function
  /// @param sender The sender of the transaction
  error IncorrectAccess(address sender);
}

// File: contracts/interfaces/IInsuranceRegistry.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "./IGrtWines.sol";

/// @title GRT Wines Insurance Registry
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @notice External registry contract for tracking if a Token (bottle of wine) has been damaged and thus should
///         not be able to be transferred to another user
/// @dev    Extended Bitmaps library is utilised to provide a gas efficent mechanism for rapidly manipulating large
//          quantities of boolean statuses. It is assumed that the structure of the Bitmap will be reliably calculated off-chain
interface IInsuranceRegistry is IGrtWines {
  //################
  //#### STRUCTS ####

  /// @notice Data structure for registering an insurance event
  /// @param firstAffectedToken The first affected token, this allows us to easily set each bucket'
  /// @param affectedTokens Bitmap of tokens that are void
  struct InsuranceEvent {
    uint256 firstAffectedToken;
    uint256[] affectedTokens;
  }

  //################
  //#### EVENTS ####
  event InsuranceEventRegistered(
    uint256 firstAffectedToken,
    uint256[] affectedTokens
  );

  //###################
  //#### FUNCTIONS ####

  /// @notice Create an insurance event
  /// @dev It is assumed that the bitmap has been adequately generated off-chain
  /// @dev Emits InsuranceEventRegistered
  /// @param insuranceEvent Insurance event data
  function createInsuranceEvent(InsuranceEvent calldata insuranceEvent)
    external;

  /// @notice Check if a token has an insurance event registered
  /// @param _tokenId The token ID to check
  /// @return isTokenAffected If TRUE token has an insurance claim - transfers except to a RedemptionManager should revert.
  function checkTokenStatus(uint256 _tokenId)
    external
    view
    returns (bool isTokenAffected);

  //################################
  //#### AUTO-GENERATED GETTERS ####
  function PLATFORM_ADMIN_ROLE() external returns (bytes32 role);
}

// File: contracts/interfaces/IListing.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "./IGrtWines.sol";

interface IListing is IGrtWines {
  //################
  //#### STRUCTS ####

  /// @dev Parameters for creating a Listing. Releases and Listings are matched with {listingId}
  /// @param releaseId The release identifier - Counter from DropManager contract
  /// @param listingId The release identifier - Counter from DropManager contract
  /// @param startDate Start date of the drop listing
  /// @param endDate End date of the drop listing
  /// @param minimumBid Minumum price to allow the listing to sell for
  /// @param startingPrice Starting price for the listing
  struct Listing {
    uint128 releaseId;
    uint40 startDate;
    uint40 endDate;
    uint256 minimumBid;
    uint256 startingPrice;
  }

  /// @param bidder The release identifier - Counter from DropManager contract
  /// @param amount  The amount of the bid
  struct Bid {
    address bidder;
    uint256 amount;
  }

  //################
  //#### EVENTS ####

  /// @dev Emitted when a Listing is created
  event ListingCreated(uint128 listingId, uint128 releaseId);

  /// @dev Emitted when a Listing is updated
  event ListingUpdated(uint128 listingId);

  /// @dev Emitted when a listing is deleted
  event ListingDeleted(uint128 listingId);

  /// @dev Emitted when a bid is successfully registered
  event BidRegistered(
    address indexed bidder,
    uint256 amount,
    uint256 tokenId,
    uint256 listingId
  );

  /// @dev Emitted when bidding is extended due to a bid being received < 10 minutes before cut-off
  event BiddingExtended(uint128 listingId);

  //################
  //#### ERRORS ####
  /// @dev Throw if listing is being deleted while it is active or completed. (TODO NEED TO UPDATE LOGIC IN ENGLISH DROP CONTRACT)
  error ListingStarted();

  /// @dev Thrown if certain operations try to be performed on already active listings
  error ListingActive();

  /// @dev Thrown if sender requests the status of a listing that was not listed at this contract.
  error NotListedHere();

  /// @dev Thrown if a bid is invalid, e.g bid < minimum bid, bid < current bid
  error InvalidBid();

  /// @dev Thrown if a bid is placed on a listing that has not started or has expired
  error ListingNotActive();

  /// @dev Thrown if validateTokenClaim or validateEthWithdrawal calls are invalid, e.g bidding still active or claimant not the bidding winner
  error InvalidClaim();

  /// @dev Thrown if attempting to distribute funds on a listings that has already had its funds distributed
  /// @param listingId The id of the listing for which the error was thrown
  error AlreadyDistributed(uint128 listingId);

  /// @dev Thrown if a distribution is attempted on a listing type that does not support it
  error DistributionNotSupported();

  //###################
  //#### FUNCTIONS ####

  /// @notice Used to create a purchase listing
  /// @dev Only callable by the dropManager (has DROP_MANAGER_ROLE)
  /// @param listingId - ID of the listing
  /// @param listing - Listing struct
  function createListing(
    uint128 listingId,
    Listing calldata listing,
    bytes calldata data
  ) external;

  /// @notice Used to update a purchase listing
  /// @dev Only callable by the dropManager (has DROP_MANAGER_ROLE)
  /// @dev Only if listing has not started yet
  /// @param listingId - ID of the listing
  /// @param listing - Listing struct
  function updateListing(
    uint128 listingId,
    Listing calldata listing,
    bytes calldata data
  ) external;

  /// @notice Used to delete a purchase listing
  /// @dev Only callable by the dropManager (has DROP_MANAGER_ROLE)
  /// @dev Only if listing has not started yet
  /// @param listingId - ID of the listing
  function deleteListing(uint128 listingId) external;

  /// @notice Utilised to register a bid for a specific token
  /// @dev Only callable by the dropManager (has DROP_MANAGER_ROLE)
  /// @dev Only if listing has not started yet
  /// @param listingId - The listing ID that this bid relates to
  /// @param tokenId - The tokenId this bid relates to
  /// @param bid - The bid itself
  function registerBid(
    uint128 listingId,
    uint256 tokenId,
    Bid calldata bid,
    bytes calldata data
  ) external;

  /// @notice Utilised to validate that a valid claim of a token is being submitted
  /// @dev Only callable by the dropManager (has DROP_MANAGER_ROLE)
  /// @dev Either returns true on success or reverts
  /// @dev  Only returns true if the claimant is the highest bidder and listing has expired
  /// @param listingId - The listing ID that this bid relates to
  /// @param tokenId - The tokenId this bid relates to
  /// @return saleAmount - The amount for which the token was sold
  function validateTokenClaim(
    uint128 listingId,
    uint128 releaseId,
    uint128 tokenId,
    address claimant
  ) external returns (uint256 saleAmount);

  /// @notice Validate whether a manual distribution is allowed for this listing. If it is valid, set the listing as distributed and return true. If not, revert
  /// @dev Only callable by the droper (has DROP_MANAGER_ROLE)
  /// @param listingId The id of the listing to validate
  /// @return valid Boolean as to whether the manual distribution is allowed
  function validateManualDistribution(uint128 listingId)
    external
    returns (bool valid);

  /// @notice Check if a listing has passed its end date
  /// @dev Should be checked before placing a bid
  /// @param listingId - The id of the listing to check
  function listingEnded(uint128 listingId) external view returns (bool status);

  //#################
  //#### GETTERS ####
  function DROP_MANAGER_ROLE() external returns (bytes32 role);
}

// File: contracts/interfaces/IRedemptionManager.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "../implementations/TokenContract.sol";
import "./IGrtWines.sol";

/// @title GrtWines Redemption Manager Contract
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @custom:contributor Seb N
/// @notice Used to manage the process of converting a Liquid Token (sitll in the Warehouse) to a Redeemed Token (Sent to the owner). User's may submit their token for redemption and the warehouse then either aborts or finalizes the redemption
/// @dev  This contract needs BURNER_ROLE on the Liquid Token contract and MINTER_ROLE on the Redeemed Token Contract
///       PLATFORM_ADMIN_ROLE and WAREHOUSE_MANAGER_ROLE are used to protect functions
///       DEFAULT_ADMIN_ROLE is not utilised for any purpose other than being the admin for all other roles
interface IRedemptionManager is IGrtWines {
  //################
  //#### STRUCTS ####

  //################
  //#### EVENTS ####
  /// @dev Emitted on successful redemption creation
  event RedemptionCreated(address indexed sender, uint256 indexed tokenId);

  /// @dev Emitted on successful redemption finalisation
  event RedemptionFinalised(address indexed sender, uint256 indexed tokenId);

  /// @dev Emitted on successful redemption finalisation
  event RedemptionAborted(address indexed sender, uint256 indexed tokenId);

  //################
  //#### ERRORS ####

  /// @dev Thrown if the sender attempts to deploy with {platformAdmin} and {superUser} set to the same address
  error AdminSuperUserMatch();

  /// @dev Thrown if the user does not posses the correct redeemable status.
  /// @param releaseId the release for which a redemption is being created.
  error RedeemableStatusIncorrect(uint256 releaseId);

  /// @dev Thrown if the user tries to set the timelock release date to before the current block time.
  /// @param releaseDate the release date which the token is redeemable from
  error ReleaseDateInvalid(uint256 releaseDate);

  //###################
  //#### FUNCTIONS ####
  /// @notice Utilised to create a redemption. Transfers the token to this contract as escrow and sets {originalOwners}
  /// @dev Account must {approveForAll} or {approve} for the specific token to redeeem
  /// @param tokenId - The token to be redeemed
  /// @param releaseId the release for which a redemption is being redeemed.
  function createRedemption(uint256 tokenId, uint256 releaseId) external;

  /// @notice Utilised to bulk finalise tokens
  /// @dev Account must have WAREHOUSE_MANAGER_ROLE to use
  /// @dev {createRedemption} must be called first. ERC721 0 address checks will fail if calling with tokens that haven't yet had a redemption created
  /// @param tokens - Array of tokens to finalise
  function finaliseRedemption(uint256[] calldata tokens) external;

  /// @notice Utilised to abort the redemption of a token
  /// @dev Account must have PLATFORM_ADMIN_ROLE to use
  /// @dev Returns the token to the original owner and deletes the value at {originalOwners}
  /// @param tokenId - Array of {FinaliseArgs} - see for more docs
  function abortRedemption(uint256 tokenId) external;

  /// @notice Utilised to set a time lock on the redemption of a token
  /// @dev Account must have PLATFORM_ADMIN_ROLE or DROP_MANAGER_ROLE to use
  /// @dev Sets the time value for a release in the timelock mapping in the Timelock contract
  /// @param releaseId - the release to update the timelock for.
  /// @param releaseDate - the date to set the timelock to.
  function setTimeLock(uint256 releaseId, uint256 releaseDate) external;

  //#################
  //#### GETTERS ####
  function PLATFORM_ADMIN_ROLE() external returns (bytes32 role);

  function WAREHOUSE_MANAGER_ROLE() external returns (bytes32 role);

  function DROP_MANAGER_ROLE() external returns (bytes32 role);

  function originalOwners(uint256 tokenId) external returns (address owner);

  function liquidToken() external returns (TokenContract implementation);

  function redeemedToken() external returns (TokenContract implementation);
}

// File: contracts/interfaces/IRoyaltyDistributor.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

interface IRoyaltyDistributor {
  /// @dev Emitted when funds are distributed to producers
  /// @param receiver The address that received the distributed funds
  /// @param amount The amount of funds received
  /// @param releaseId The id of the release for which the distribution occurred
  event FundsDistributed(address receiver, uint256 amount, uint128 releaseId);

  /// @dev Emitted when ether funds are received
  /// @param sender The sender of the funds
  /// @param amount The amount of funds received
  event EthReceived(address sender, uint256 amount);

  /// @dev Thrown if an ETH transfer fails
  /// @param destination The receiver of the ether
  /// @param amount The amount of ether being sent
  error EthTransferFailed(address destination, uint256 amount);

  /// @dev Thrown if an invalid amount is provided
  error InvalidAmount();

  /// @dev Thrown if attempting to transfer a zero eth amount
  error InvalidEthAmount();

  /// @notice Distribute funds accumulated from secondary marketplaces to producers and the GRT royalty wallet
  /// @dev Only callable by the DROP_MANAGER_ROLE
  /// @param amount The total value of the sale to be distributed
  /// @param receiver The address of the receiver of the funds percentage (i.e the producer)
  /// @param percentage The percentage of funds to distribute. Should account for decimal precision of 10**2
  /// @param royaltyWallet The address if the royalty wallet to receive remaining funds not sent to the producer
  function distributeFunds(
    uint256 amount,
    address receiver,
    uint16 percentage,
    address royaltyWallet,
    uint128 releaseId
  ) external;
}

// File: contracts/interfaces/ISwap.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;

import "@openzeppelin/contracts/token/ERC20/ERC20.sol";

interface ISwap {
  /** ERRORS */

  /// @notice reverts when a zero address is passed in as a potential admin or smart contract location
  error CannotBeZeroAddress();
  /// @notice reverts when the 0x token swap fails
  error ZeroXSwapFailed();
  /// @notice reverts when the user provides an incorrect buy token
  error IncorrectBuyToken();

  /**
    @notice emits when a user swaps a token to deposit into a thin wallet
    @param user the user who swapped the token
    @param amount the amount the user swapped of their token
    @param token the token that the user swapped
    @param target the thin wallet that the user was depositing into
  */
  event SwapDeposit(address user, uint256 amount, ERC20 token, address target);

  /**
    @notice emits when a user swaps a token to deposit into a thin wallet
    @param user the user who swapped the token
    @param amount the amount the user swapped of their token
    @param target the thin wallet that the user was depositing into
  */
  event SwapETH(address user, uint256 amount, address target);

  /**
    @notice this function is used to swap donated ETH to a desired token before it is deposited into another external contract
    @param buyToken the token that the user wishes to swap for, should be the base token
    @param amount the amount the user is selling of their sellToken
    @param location where the bought tokens will be sent after they are swapped
    @param spender an address provided by the 0x Quote API
    @param swapTarget an address provided by the 0x Quote API
    @param swapCallData the calldata provided by the 0x Quote API
  */
  function depositETH(
    ERC20 buyToken,
    uint256 amount,
    address location,
    address spender,
    address payable swapTarget,
    bytes calldata swapCallData
  ) external payable;

  /**
    @notice this function is used to swap donated ERC20 tokens to a desired token before it is deposited into another external contract
    @param sellToken the token that the user is selling to acquire the base token
    @param buyToken the token that the user wishes to swap for, should be the base token
    @param amount the amount the user is selling of their sellToken
    @param location where the bought tokens will be sent after they are swapped
    @param spender an address provided by the 0x Quote API
    @param swapTarget an address provided by the 0x Quote API
    @param swapCallData the calldata provided by the 0x Quote API
  */
  function depositERC20(
    ERC20 sellToken,
    ERC20 buyToken,
    uint256 amount,
    address location,
    address spender,
    address payable swapTarget,
    bytes calldata swapCallData
  ) external payable;

  function baseToken() external view returns (ERC20);
}
// File: contracts/interfaces/ITokenContract.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "../interfaces/IInsuranceRegistry.sol";
import "../interfaces/ITokenRegistry.sol";
import "./IGrtWines.sol";

/// @title TokenContract
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @custom:contributor mfbevan (mfbevan.eth)
/// @custom:contributor Brodie S
/// @notice Implementation to be used for the Liquid and Redeemed editions of each token
/// @dev  The GRT Wines architecture uses a dual ERC721 token system. When releases are created the `LiquidToken`
///       is minted and can be purchased through various listing mechanisms via the Drop Manager. When a user
///       wishes to redeem their token for a physical asset, the `LiquidToken` is burned, and a `RedeemedToken` is
///       minted. The same `TokenContract` implementation is deployed twice, once for each edition. The metadata
///       for both the Liquid and Redeemed editions of each token is set when a release is created, and manage by
///       the `TokenRegistry`. Almost entirely stock ERC721 with the exception of externalised mint, burn and update
///       token URI functions which will be guarded by Open Zeppelin RBAC.
///       DEFAULT_ADMIN_ROLE is not utilised for any purpose other than being the admin for all other roles
interface ITokenContract is IGrtWines, IERC721 {
  //################
  //#### STRUCTS ####

  /// @dev Holds the arguments for a mint transaction
  /// @param to The account the token should be minted to
  struct MintArgs {
    address to;
  }

  /// @dev Holds the arguments necessary for minting tokens with a specific ID
  /// @param to The token ID to be locked
  /// @param tokenId The token ID to be minted
  struct MintWithIdArgs {
    address to;
    uint256 tokenId;
  }

  //################
  //#### ERRORS ####

  /// @dev Thrown if a transaction attempts to update the metadata for a token that has already had an update (locked)
  /// @param sender The sender of the transaction
  /// @param tokenId The tokenId that resulted in the error
  error TokenLocked(address sender, uint256 tokenId);

  /// @dev Thrown if an account attempts to transfer a token that has an insurance event AND msg.sender != redemptionManager
  /// @param tokenId The token ID the attempted to be transferred
  error InsuranceEventRegistered(uint256 tokenId);

  //###################
  //#### FUNCTIONS ####

  /// @notice External mint funciton for Tokens
  /// @dev Bulk mint one or more tokens via MintArgs array for gas efficency.
  /// @dev Only accessible to PLATFORM_ADMIN_ROLE or MINTER_ROLE
  /// @param receiver The address to receive the minted NFTs. This should be the DropManager
  /// @param qty The number of tokens to mint
  /// @param liquidUri The liquid token URI to set for the batch
  /// @param redeemedUri The redeemed token URI to set for the batch
  function mint(
    address receiver,
    uint128 qty,
    string memory liquidUri,
    string memory redeemedUri
  ) external returns (uint256 mintCount);

  /// @notice External mint function to allow minting token with an explicit ID
  /// @dev Bulk mint one or more tokens with an explicit ID - intended to be used by the RedemptionManager to maintain
  /// @dev Only accessible to MINTER_ROLE which should only be assigned to the RedemptionManager when this contract is deployed as the RedeemedToken
  /// @dev This does not set the metadata as it is assumed that the metadata will already have been set in the TokenRegistry on mint of the Liquid Token
  /// @param mintWithIdArgs - Array of MintWithIdArgs struct. See {MintWithIdArgs} for param docs
  function mintWithId(MintWithIdArgs[] calldata mintWithIdArgs) external;

  /// @notice External burn funciton
  /// @dev Only accessible to PLATFORM_ADMIN_ROLE or BURNER_ROLE
  /// @param tokens - Array of token IDs to burn
  function burn(uint256[] calldata tokens) external;

  /// @notice Change the metadata URI for a given token batch
  /// @dev Tokens may only be updated once
  /// @dev Only accessible to PLATFORM_ADMIN_ROLE
  /// @param batchIndex The index of the batch to update in the tokenKeys array
  /// @param liquidUri The new liquid token URI to set
  /// @param redeemedUri The new redeemed token URI to set
  function changeTokenMetadata(
    uint256 batchIndex,
    string memory liquidUri,
    string memory redeemedUri
  ) external;

  /// @notice Lock the capability for a token to be updated
  /// @dev This behaves like a fuse and cannot be undone
  /// @dev Only accessible to PLATFORM_ADMIN_ROLE
  /// @param batchIndex The index of the batch to lock
  function lockTokenMetadata(uint256 batchIndex) external;

  /// @notice Set the insurance registry address
  /// @param _registryAddress The Address of the insurance registry
  function setInsuranceRegistry(address _registryAddress) external;

  /// @notice Set the redemption manager address
  /// @param _managerAddress The address of the redemption manager
  function setRedemptionManager(address _managerAddress) external;

  /// @notice Set the address and percentage of secondary market fees
  /// @param receiver The receiver wallet for the secondary market fees. This should be the address of the Royalty Distributor
  /// @param feeNumerator The fee percentage to send to the distributor, expressed in basis points
  function setSecondaryRoyalties(address receiver, uint96 feeNumerator)
    external;

  //#################
  //#### GETTERS ####

  function PLATFORM_ADMIN_ROLE() external returns (bytes32 role);

  function MINTER_ROLE() external returns (bytes32 role);

  function BURNER_ROLE() external returns (bytes32 role);

  /// @dev returns the locked status from the TokenRegistry
  function tokenLocked(uint256 tokenId) external view returns (bool hasUpdated);

  function insuranceRegistry()
    external
    returns (IInsuranceRegistry registryAddress);

  function tokenRegistry()
    external
    returns (ITokenRegistry tokenRegistryAddress);

  function redemptionManager() external returns (address managerAddress);
}

// File: contracts/interfaces/ITokenRegistry.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

import "./IGrtWines.sol";

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

/// @title Token Registry
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor mfbevan (mfbevan.eth)
/// @notice Stores the token URIs in batches allowing for an arbitrary number of tokens to be minted at a time
interface ITokenRegistry is IGrtWines {
  //################
  //#### STRUCTS ####

  /// @dev Parameters for the Token URIs in a release
  /// @param liquidUri The base uri for the liquid token metadata
  /// @param redeemedUri The base uri for the liquid token metadata
  /// @param owner The current owner of the token
  /// @param key The id of the first token in the batch
  /// @param count The number of tokens in the batch
  /// @param locked Is this token batch locked from having its token uri being set
  struct TokenKey {
    string liquidUri;
    string redeemedUri;
    uint256 key;
    uint16 count;
    bool locked;
  }

  //################
  //#### EVENTS ####

  /// @dev Emitted when a token URI is successfully updated
  /// @param batchIndex The index of the batch that was updated
  /// @param liquidUri The updated liquid token URI
  /// @param redeemedUri The updated redeemed token URI
  event BatchURIUpdated(
    uint256 indexed batchIndex,
    string liquidUri,
    string redeemedUri
  );

  /// @dev Emitted when token metadata is successfully locked
  /// @param batchIndex The index of the batch that was locked
  event BatchMetadataLocked(uint256 indexed batchIndex);

  //################
  //#### ERRORS ####

  /// @dev Thrown if a transaction attempts to update the metadata for a batch that has already had an update (locked)
  /// @param batchId The id of the batch being queried
  error BatchLocked(uint256 batchId);

  /// @dev Thrown if querying a batch index that does not exist yet
  error InvalidBatchIndex();

  /// @dev Thrown if batch metadata is added in a non-consecutive order
  error InvalidBatchData();

  /// @dev Thrown if searching for a token that does not exist in the TokenRegistry
  error InvalidTokenId();

  //###################
  //#### FUNCTIONS ####

  /// @notice Get the token key corresponding to a token
  /// @dev If the owner of the token in the owners mapping is the zero address, return the address of the DropManager
  /// @param _tokenId Id of the token
  /// @return tokenKey - the token key containing the liquid and redeemed token URIs
  function getTokenKey(uint _tokenId)
    external
    view
    returns (TokenKey memory tokenKey);

  /// @notice Add a new metadata batch
  /// @dev New batch will be pushed to the end of the tokenKeys array
  /// @dev Only accessible to TOKEN_CONTRACT_ROLE
  /// @dev Emites a {BatchURIUpdated} event
  /// @param _tokenKey The new token batch to add
  function addBatchMetadata(TokenKey calldata _tokenKey) external;

  /// @notice Update the metadata URI for a token batch
  /// @dev Token URIs may only be updated once, and this function will call {lockBatchMetadata}
  /// @dev Only accessible to TOKEN_CONTRACT_ROLE
  /// @dev Emits a {BatchURIUpdated} event
  /// @param _batchIndex The index of the batch to update in the tokenKeys array
  /// @param _liquidUri The new liquid token URI to set
  /// @param _redeemedUri The new redeemed token URI to set
  function updateBatchMetadata(
    uint256 _batchIndex,
    string calldata _liquidUri,
    string calldata _redeemedUri
  ) external;

  /// @notice Lock the capability for a batch to be updated
  /// @dev This behaves like a fuse and cannot be undone
  /// @dev Emits a {BatchMetadataLocked} event
  /// @dev Only accessible to TOKEN_CONTRACT_ROLE
  /// @param _batchIndex The index of the batch to lock
  function lockBatchMetadata(uint256 _batchIndex) external;

  //#################
  //#### GETTERS ####

  function TOKEN_CONTRACT_ROLE() external returns (bytes32 role);
}

// File: contracts/libraries/GrtLibrary.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/// @title A title that should describe the contract/interface
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor mfbevan (mfbevan.eth)
/// @notice Helpers for GRT Wines contracts
library GrtLibrary {
  /// @dev Thrown whenever a zero-address check fails
  /// @param field The name of the field on which the zero-address check failed
  error ZeroAddress(bytes32 field);

  /// @notice Check if a field is the zero address, if so revert with the field name
  /// @param _address The address to check
  /// @param _field The name of the field to check
  function checkZeroAddress(address _address, bytes32 _field) internal pure {
    if (_address == address(0)) {
      revert ZeroAddress(_field);
    }
  }
}

// File: contracts/vendors/ExtendedBitmap.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

/*
  _______ .______     .___________.____    __    ____  __  .__   __.  _______     _______.
 /  _____||   _  \    |           |\   \  /  \  /   / |  | |  \ |  | |   ____|   /       |
|  |  __  |  |_)  |   `---|  |----` \   \/    \/   /  |  | |   \|  | |  |__     |   (----`
|  | |_ | |      /        |  |       \            /   |  | |  . `  | |   __|     \   \    
|  |__| | |  |\  \----.   |  |        \    /\    /    |  | |  |\   | |  |____.----)   |   
 \______| | _| `._____|   |__|         \__/  \__/     |__| |__| \__| |_______|_______/    
                                                                                     
*/

/// @title Extended Bitmaps
/// @author Developed by Labrys on behalf of GRT Wines
/// @custom:contributor Sean L (slongdotexe)
/// @dev    Extended Bitmaps library is utilised to provide a gas efficent mechanism for rapidly manipulating large
//          quantities of boolean statuses. It is assumed that the structure of the Bitmap will be reliably calculated off-chain
library ExtendedBitmap {
  struct BitMap {
    mapping(uint256 => uint256) _data;
  }

  function getBucket(BitMap storage bitmap, uint256 bucketIndex)
    internal
    view
    returns (uint256 _bucket)
  {
    _bucket = bitmap._data[bucketIndex];
  }

  function setBucket(
    BitMap storage bitmap,
    uint256 bucketIndex,
    uint256 bucketContents
  ) internal {
    bitmap._data[bucketIndex] = bucketContents;
  }

  function getMaskedBucket(
    BitMap storage bitmap,
    uint256 startIndex,
    uint256 endIndex,
    uint256 index
  ) internal view returns (uint256) {
    uint256 startBucket = startIndex >> 8;
    uint256 endBucket = endIndex >> 8;
    uint256 currentBucket = index >> 8;
    uint256 result = getBucket(bitmap, currentBucket);
    //If the currentBucket we're accessing is the first bucket for this range of bits, mask the first N bits
    if (currentBucket == startBucket) {
      //The number of bits we want to mask off the start of the word
      uint256 maskNStarting = startIndex - 256 * startBucket;
      result &= type(uint256).max << maskNStarting;
    }
    //If the currentBucket we're accessing is the last bucket for this range of bits, mask the last N bits
    if (currentBucket == endBucket) {
      //The number of bits we want to mask off the end of the word
      uint256 maskNEnding = 255 - (endIndex - 256 * endBucket);
      result &= type(uint256).max >> maskNEnding;
    }
    return result;
  }

  // ################################################################
  // ## ALL BELOW FUNCTIONS ARE DIRECTLY COPIED FROM OPEN-ZEPPELIN ##
  // ################################################################

  /**
   * @dev Returns whether the bit at `index` is set.
   */
  function get(BitMap storage bitmap, uint256 index)
    internal
    view
    returns (bool)
  {
    uint256 bucket = index >> 8;
    uint256 mask = 1 << (index & 0xff);
    return bitmap._data[bucket] & mask != 0;
  }

  /**
   * @dev Sets the bit at `index` to the boolean `value`.
   */
  function setTo(
    BitMap storage bitmap,
    uint256 index,
    bool value
  ) internal {
    if (value) {
      set(bitmap, index);
    } else {
      unset(bitmap, index);
    }
  }

  /**
   * @dev Sets the bit at `index`.
   */
  function set(BitMap storage bitmap, uint256 index) internal {
    uint256 bucket = index >> 8;
    uint256 mask = 1 << (index & 0xff);
    bitmap._data[bucket] |= mask;
  }

  /**
   * @dev Unsets the bit at `index`.
   */
  function unset(BitMap storage bitmap, uint256 index) internal {
    uint256 bucket = index >> 8;
    uint256 mask = 1 << (index & 0xff);
    bitmap._data[bucket] &= ~mask;
  }
}

// File: operator-filter-registry/src/DefaultOperatorFilterer.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

import {OperatorFilterer} from "./OperatorFilterer.sol";
import {CANONICAL_CORI_SUBSCRIPTION} from "./lib/Constants.sol";
/**
 * @title  DefaultOperatorFilterer
 * @notice Inherits from OperatorFilterer and automatically subscribes to the default OpenSea subscription.
 * @dev    Please note that if your token contract does not provide an owner with EIP-173, it must provide
 *         administration methods on the contract itself to interact with the registry otherwise the subscription
 *         will be locked to the options set during construction.
 */

abstract contract DefaultOperatorFilterer is OperatorFilterer {
    /// @dev The constructor that is called when the contract is being deployed.
    constructor() OperatorFilterer(CANONICAL_CORI_SUBSCRIPTION, true) {}
}

// File: operator-filter-registry/src/IOperatorFilterRegistry.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

interface IOperatorFilterRegistry {
    /**
     * @notice Returns true if operator is not filtered for a given token, either by address or codeHash. Also returns
     *         true if supplied registrant address is not registered.
     */
    function isOperatorAllowed(address registrant, address operator) external view returns (bool);

    /**
     * @notice Registers an address with the registry. May be called by address itself or by EIP-173 owner.
     */
    function register(address registrant) external;

    /**
     * @notice Registers an address with the registry and "subscribes" to another address's filtered operators and codeHashes.
     */
    function registerAndSubscribe(address registrant, address subscription) external;

    /**
     * @notice Registers an address with the registry and copies the filtered operators and codeHashes from another
     *         address without subscribing.
     */
    function registerAndCopyEntries(address registrant, address registrantToCopy) external;

    /**
     * @notice Unregisters an address with the registry and removes its subscription. May be called by address itself or by EIP-173 owner.
     *         Note that this does not remove any filtered addresses or codeHashes.
     *         Also note that any subscriptions to this registrant will still be active and follow the existing filtered addresses and codehashes.
     */
    function unregister(address addr) external;

    /**
     * @notice Update an operator address for a registered address - when filtered is true, the operator is filtered.
     */
    function updateOperator(address registrant, address operator, bool filtered) external;

    /**
     * @notice Update multiple operators for a registered address - when filtered is true, the operators will be filtered. Reverts on duplicates.
     */
    function updateOperators(address registrant, address[] calldata operators, bool filtered) external;

    /**
     * @notice Update a codeHash for a registered address - when filtered is true, the codeHash is filtered.
     */
    function updateCodeHash(address registrant, bytes32 codehash, bool filtered) external;

    /**
     * @notice Update multiple codeHashes for a registered address - when filtered is true, the codeHashes will be filtered. Reverts on duplicates.
     */
    function updateCodeHashes(address registrant, bytes32[] calldata codeHashes, bool filtered) external;

    /**
     * @notice Subscribe an address to another registrant's filtered operators and codeHashes. Will remove previous
     *         subscription if present.
     *         Note that accounts with subscriptions may go on to subscribe to other accounts - in this case,
     *         subscriptions will not be forwarded. Instead the former subscription's existing entries will still be
     *         used.
     */
    function subscribe(address registrant, address registrantToSubscribe) external;

    /**
     * @notice Unsubscribe an address from its current subscribed registrant, and optionally copy its filtered operators and codeHashes.
     */
    function unsubscribe(address registrant, bool copyExistingEntries) external;

    /**
     * @notice Get the subscription address of a given registrant, if any.
     */
    function subscriptionOf(address addr) external returns (address registrant);

    /**
     * @notice Get the set of addresses subscribed to a given registrant.
     *         Note that order is not guaranteed as updates are made.
     */
    function subscribers(address registrant) external returns (address[] memory);

    /**
     * @notice Get the subscriber at a given index in the set of addresses subscribed to a given registrant.
     *         Note that order is not guaranteed as updates are made.
     */
    function subscriberAt(address registrant, uint256 index) external returns (address);

    /**
     * @notice Copy filtered operators and codeHashes from a different registrantToCopy to addr.
     */
    function copyEntriesOf(address registrant, address registrantToCopy) external;

    /**
     * @notice Returns true if operator is filtered by a given address or its subscription.
     */
    function isOperatorFiltered(address registrant, address operator) external returns (bool);

    /**
     * @notice Returns true if the hash of an address's code is filtered by a given address or its subscription.
     */
    function isCodeHashOfFiltered(address registrant, address operatorWithCode) external returns (bool);

    /**
     * @notice Returns true if a codeHash is filtered by a given address or its subscription.
     */
    function isCodeHashFiltered(address registrant, bytes32 codeHash) external returns (bool);

    /**
     * @notice Returns a list of filtered operators for a given address or its subscription.
     */
    function filteredOperators(address addr) external returns (address[] memory);

    /**
     * @notice Returns the set of filtered codeHashes for a given address or its subscription.
     *         Note that order is not guaranteed as updates are made.
     */
    function filteredCodeHashes(address addr) external returns (bytes32[] memory);

    /**
     * @notice Returns the filtered operator at the given index of the set of filtered operators for a given address or
     *         its subscription.
     *         Note that order is not guaranteed as updates are made.
     */
    function filteredOperatorAt(address registrant, uint256 index) external returns (address);

    /**
     * @notice Returns the filtered codeHash at the given index of the list of filtered codeHashes for a given address or
     *         its subscription.
     *         Note that order is not guaranteed as updates are made.
     */
    function filteredCodeHashAt(address registrant, uint256 index) external returns (bytes32);

    /**
     * @notice Returns true if an address has registered
     */
    function isRegistered(address addr) external returns (bool);

    /**
     * @dev Convenience method to compute the code hash of an arbitrary contract
     */
    function codeHashOf(address addr) external returns (bytes32);
}

// File: operator-filter-registry/src/OperatorFilterer.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

import {IOperatorFilterRegistry} from "./IOperatorFilterRegistry.sol";
import {CANONICAL_OPERATOR_FILTER_REGISTRY_ADDRESS} from "./lib/Constants.sol";
/**
 * @title  OperatorFilterer
 * @notice Abstract contract whose constructor automatically registers and optionally subscribes to or copies another
 *         registrant's entries in the OperatorFilterRegistry.
 * @dev    This smart contract is meant to be inherited by token contracts so they can use the following:
 *         - `onlyAllowedOperator` modifier for `transferFrom` and `safeTransferFrom` methods.
 *         - `onlyAllowedOperatorApproval` modifier for `approve` and `setApprovalForAll` methods.
 *         Please note that if your token contract does not provide an owner with EIP-173, it must provide
 *         administration methods on the contract itself to interact with the registry otherwise the subscription
 *         will be locked to the options set during construction.
 */

abstract contract OperatorFilterer {
    /// @dev Emitted when an operator is not allowed.
    error OperatorNotAllowed(address operator);

    IOperatorFilterRegistry public constant OPERATOR_FILTER_REGISTRY =
        IOperatorFilterRegistry(CANONICAL_OPERATOR_FILTER_REGISTRY_ADDRESS);

    /// @dev The constructor that is called when the contract is being deployed.
    constructor(address subscriptionOrRegistrantToCopy, bool subscribe) {
        // If an inheriting token contract is deployed to a network without the registry deployed, the modifier
        // will not revert, but the contract will need to be registered with the registry once it is deployed in
        // order for the modifier to filter addresses.
        if (address(OPERATOR_FILTER_REGISTRY).code.length > 0) {
            if (subscribe) {
                OPERATOR_FILTER_REGISTRY.registerAndSubscribe(address(this), subscriptionOrRegistrantToCopy);
            } else {
                if (subscriptionOrRegistrantToCopy != address(0)) {
                    OPERATOR_FILTER_REGISTRY.registerAndCopyEntries(address(this), subscriptionOrRegistrantToCopy);
                } else {
                    OPERATOR_FILTER_REGISTRY.register(address(this));
                }
            }
        }
    }

    /**
     * @dev A helper function to check if an operator is allowed.
     */
    modifier onlyAllowedOperator(address from) virtual {
        // Allow spending tokens from addresses with balance
        // Note that this still allows listings and marketplaces with escrow to transfer tokens if transferred
        // from an EOA.
        if (from != msg.sender) {
            _checkFilterOperator(msg.sender);
        }
        _;
    }

    /**
     * @dev A helper function to check if an operator approval is allowed.
     */
    modifier onlyAllowedOperatorApproval(address operator) virtual {
        _checkFilterOperator(operator);
        _;
    }

    /**
     * @dev A helper function to check if an operator is allowed.
     */
    function _checkFilterOperator(address operator) internal view virtual {
        // Check registry code length to facilitate testing in environments without a deployed registry.
        if (address(OPERATOR_FILTER_REGISTRY).code.length > 0) {
            // under normal circumstances, this function will revert rather than return false, but inheriting contracts
            // may specify their own OperatorFilterRegistry implementations, which may behave differently
            if (!OPERATOR_FILTER_REGISTRY.isOperatorAllowed(address(this), operator)) {
                revert OperatorNotAllowed(operator);
            }
        }
    }
}

// File: operator-filter-registry/src/lib/Constants.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.13;

address constant CANONICAL_OPERATOR_FILTER_REGISTRY_ADDRESS = 0x000000000000AAeB6D7670E522A718067333cd4E;
address constant CANONICAL_CORI_SUBSCRIPTION = 0x3cc6CddA760b79bAfa08dF41ECFA224f810dCeB6;